

Conventional Programming

No.	Title	PGM-No.
Drilling		
1	Two rows of holes	7280
2	Irregular drilling pattern, 3 tools	7153
3	Linear hole pattern	7109
4	Linear offset hole pattern	7426
5	Four double bolt hole circles, 3 tools	7139
6	Large drilled plate	7411
2D Milling		
7	Loading frame	72810
8	Double L-form with 2D-series cycles	7172
9	Mixed elements (internal) with Q-parameters	761541
10	Mixed elem. (internal and external) with Q-param.	761591
Free Contour Programming		
11	FK Telephone	75103
12	FK Dumbbell	75116
13	FK Hammer basic	75119
14	FK Hammer, island with 20-series cycles	7173
15	FK Mickey Mouse basic	71750
16	FK Mickey Mouse with 20-series cycles	7175
17	FK T-Contour with 20-series cycles	7174
18	FK Comb, basic, uncorrected	75108
19	FK Comb with 20-series cycles	7176
20	FK Comb first original, then mirrored in Y	75122
„New“ Functions		
21	Multiple use of an island	7177


Contents: Advanced course TNC 4xx

No.	Title	PGM-No.
	3D Milling	
22	Toggle joint, 1 tool	761260
23	Toggle joint, 2 tools	761261
24	Truncated cone, standing, 2D, external, horizontal	7162
25	Truncated cone, standing, 3D, external	7282
26	Truncated cone, internal, 3D, basic	7283
27	Truncated cone, internal, 3D, from solid block	7284
28	Hemisphere, internal, 2D, horizontal	7322
29	Hemisphere, internal, 2D, horiz. with pockets	7323
30	Hemisphere, external, 2D, horizontal	72841
31	Hemisphere, external, 3D, vertical, 1 cut	76130
32	Hemisphere, external, 3D, vertical, 2 cuts	76131
33	Threadmilling in three steps, 2 settings	7164

Q-Parameter Programming

No.	Title	PGM-No.
	Milling	
34	Milling a contour with several Z-settings	7285
35	Finish milling rectangular pocket	7286
36	Milling cylindrical pins	7288
	Drilling	
37	1 Simple bolt hole circle	7416
38	6 Bolt hole circles, 1 tool	74160
39	6 Bolt hole circles, 3 tools, 3 cycles	74161
40	Linear hole pattern	746910





Program layout:

Conventional preparation

```

BLK FORM
TOOL DEF
TOOL CALL 1 Z S3000

CYCL DEF 1.0 PECKING

L X... Y... R0 F9999 M3
L Z...

① CALL LBL 1

L X... Y...
L Z...

② CALL LBL 1
    
```

Workpiece blank

First tool call

Pecking cycle

Pre-position

Pre-position

Retract tool, end

```
L Z... M2
```


SPGM 1

```

LBL 1
L IX+20 M99
CALL LBL 1 REP 3/3
LBL 0
    
```

Solution:

Two rows of holes



Main program

```
0 BEGIN PGM 7280 MM
1 ..... TWO ROWS OF HOLES
2 BLK FORM 0.1 Z X+0 Y+0 Z-30
3 BLK FORM 0.2 X+100 Y+60 Z+0
4 TOOL DEF 1 L+0 R+2,5
5 TOOL CALL 1 Z S3000
6 CYCL DEF 1.0 PECKING
7 CYCL DEF 1.1 SET UP -2
8 CYCL DEF 1.2 DEPTH -22
9 CYCL DEF 1.3 PECKG -10
10 CYCL DEF 1.4 DWELL 0
11 CYCL DEF 1.5 F150

12 L X+5 Y+45 R0 F9999 M3
13 L Z+2
14 CALL LBL 1

15 L X+10 Y+15
16 CALL LBL 1

Retract tool, end 17 L Z+20 R0 F9999 M2
```

SPGM 1

```
18 LBL 1
19 L IX+20 F9999 M99
20 CALL LBL 1 REP 3/3
21 LBL 0
22 END PGM 7280 MM
```



HEIDENHAIN

Advanced course TNC 4xx



C04



7280/2

Task: **Irregular drilling pattern,
with 3 tools and 3 cycles**

Program(s): _____



- Tools:**
- NC-center drill
 - Twist drill
 - Tap

Program layout: **Irregular drilling pattern,
with 3 tools and 3 cycles**

Preparation

```
BLK FORM
TOOL DEF
TOOL 1...

L Z... M6
```

Workpiece blank
Tool list
First tool call

Tool change

**Process
Center**

```
CYCL DEF, DEPTH = PECKG
CALL LBL 1
STOP M6
```

Pecking cycle

Call drilling pattern

Tool change

Pecking

```
TOOL 2...
CYCL DEF, DEPTH ≠ PECKG
CALL LBL 1
STOP M6
```

Pecking cycle

Call drilling pattern

Tool change

Tapping

```
TOOL 3...
CYCL DEF
CALL LBL 1
```

Tapping cycle

Call drilling pattern

Retract tool, end

```
L Z... M2
```

**Drilling pattern,
SPGM 1**

```
LBL 1
L X... Y...
L Z... M89

L X... Y...

L X... Y... M99

L Z...
L X... Y...

LBL 0
```

1st position lower left
Setup clearance
and modal cycle call

Further hole positions:
automatic

Further hole positions:
Cycle call by block

Tool change position

Program layout:

Use of Q-parameter for irregular drilling pattern

BLK-FORM

Center

```
TOOL DEF 1 L0 R3,5  
TOOL CALL ...  
Q1 = +2  
Q2 = +3  
  
CYCL DEF ... PECKING  
SET UP -Q1  
DEPTH -Q2 ...
```

Set up
Depth

```
CALL LBL 1
```

Pecking

```
TOOL DEF 2 L0 R2,5  
TOOL CALL ...  
Q2 = 20
```

Cycle type stays the
same!

```
CALL LBL 1
```

Tapping

```
TOOL DEF 3 L0 R3,0  
TOOL CALL ...  
Q1 = 6  
Q2 = 15  
CYCL DEF ... TAPPING  
SET UP -Q1  
DEPTH -Q2 ...
```

```
CALL LBL 1
```

Retract tool, end

```
L Z100 M2
```

Return jump

**Drilling pattern,
SPGM**

```
LBL 1  
  
L X... Y... R0 F9999 M3  
L ZQ1 M89  
L X... Y...  
L X... Y... M99  
L Z50  
  
LBL 0
```



HEIDENHAIN

Advanced course TNC 4xx


C04



G7

Solution:

Irregular drilling pattern, with 3 tools and 3 cycles



Main program

```

0 BEGIN PGM 7153 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5 ..... NC-CENTER DRILL
4 TOOL DEF 2 L+0 R+2,5 ..... TWIST DRILL
5 TOOL DEF 3 L+0 R+3 ..... TAP

```

Center

```

6 TOOL CALL 1 Z S800 ..... NC-CENTER DRILL
7 L Z+100 R0 F9999 M6 ..... TOOL CHANGE
8 CYCL DEF 1.0 PECKING
9 CYCL DEF 1.1 SET UP -2
10 CYCL DEF 1.2 DEPTH -2
11 CYCL DEF 1.3 PECKG -2
12 CYCL DEF 1.4 DWELL 0
13 CYCL DEF 1.5 F200
14 CALL LBL 1 ..... CALL DRILLING PATTERN
15 STOP M6 ..... TOOL CHANGE

```

Pecking

```

16 TOOL CALL 2 Z S500 ..... TWIST DRILL
17 CYCL DEF 1.0 PECKING
18 CYCL DEF 1.1 SET UP -2
19 CYCL DEF 1.2 DEPTH -20
20 CYCL DEF 1.3 PECKG -10
21 CYCL DEF 1.4 DWELL 0
22 CYCL DEF 1.5 F100
23 CALL LBL 1 ..... CALL DRILLING PATTERN
24 STOP M6 ..... TOOL CHANGE

```

Tapping

```

25 TOOL CALL 3 Z S500 ..... TAP
26 CYCL DEF 2.0 TAPPING
27 CYCL DEF 2.1 SET UP -2
28 CYCL DEF 2.2 DEPTH -15
29 CYCL DEF 2.3 DWELL 0
30 CYCL DEF 2.4 F500
31 CALL LBL 1 ..... CALL DRILLING PATTERN

```

Retract tool, end

```

32 L Z+100 M2

```



HEIDENHAIN

Advanced course TNC 4xx

C01



7153/3

Solution:


Irregular drilling pattern, with 3 tools and 3 cycles

SPGM 1, Drilling pattern

```
33 LBL 1
34 L X+10 Y+20 R0 F9999 M13
35 L Z+2 M89 ..... FIRST MODAL CYCLE CALL
36 L X+40 Y+30
37 L X+90 Y+10
38 L Y+40
39 L X+50 Y+80
40 L Y+50
41 L X+30 Y+40
42 L X+20 Y+80
43 L X+10 Y+90
44 L X+70 M99 ..... LAST CYCLE CALL BY BLOCK
45 L Z+100 M9 ..... TOOL CHANGE POSITION
46 L X-20 Y-20
47 LBL 0
48 END PGM 7153 MM
```

Task: **Linear hole pattern**

Program(s): _____



Procedure: • Move up and down columns (meandering)

Tools:

- NC-center drill
- Twist drill
- Tap

Program layout: **Linear hole pattern**

Preparation

```
BLK FORM  
TOOL 1.../TOOL 2...
```

Workpiece blank
Tool list if necessary

**Process
Center**

```
TOOL 1...  
CYCL DEF, DEPTH = PECKG
```

Pecking cycle

```
CALL LBL 1
```

Call drilling pattern

```
L Z... M6
```

Tool change

Pecking

```
TOOL 2...  
CYCL DEF, DEPTH ≠ PECKG
```

Pecking cycle

```
CALL LBL 1
```

Call drilling pattern

```
L Z... M6
```

Tool change

Tapping

```
TOOL 3...  
CYCL DEF
```

Tapping cycle

```
CALL LBL 1
```


Call drilling pattern

Retract tool, end

```
L Z... M2
```



Meandering movement:



Drilling pattern, SPM 1

<i>LBL 1</i>	
<i>L X... Y... R0 F9999 M3</i> <i>L Z... M8</i>	
<i>LBL 2</i>	
<i>L IX... M99</i>	
<i>LBL 3</i>	
<i>L IY... M99</i>	
<i>CALL LBL 3 REP...</i>	
<i>L IX... M99</i>	
<i>LBL 4</i>	
<i>L IY... M99</i>	
<i>CALL LBL 4 REP...</i>	
<i>CALL LBL 2 REP...</i>	
<i>LBL 0</i>	

Absolute Pre-position
Setup clearance

Label
Define increments
Cross over and drill

Label
Move up column


Cross over and drill

Label
Move down column

Remaining columns

Solution:

Linear hole pattern



Main program

```

0 BEGIN PGM 7109 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5 ..... NC-CENTER DRILL
4 TOOL DEF 2 L+0 R+2,5 ..... TWIST DRILL
5 TOOL DEF 3 L+0 R+3 ..... TAP

```

Center

```

6 TOOL CALL 1 Z S2500 ..... NC-CENTER DRILL
7 L Z+100 R0 F9999 M6 ..... TOOL CHANGE
8 CYCL DEF 1.0 PECKING
9 CYCL DEF 1.1 SET UP -2
10 CYCL DEF 1.2 DEPTH -3
11 CYCL DEF 1.3 PECKG -3
12 CYCL DEF 1.4 DWELL 0
13 CYCL DEF 1.5 F100
14 CALL LBL 1 ..... CALL DRILLING PATTERN
15 L Z+100 M6 ..... TOOL CHANGE

```

Pecking

```

16 TOOL CALL 2 Z S1500
17 CYCL DEF 1.0 PECKING
18 CYCL DEF 1.1 SET UP -2
19 CYCL DEF 1.2 DEPTH -20
20 CYCL DEF 1.3 PECKG -20
21 CYCL DEF 1.4 DWELL 0
22 CYCL DEF 1.5 F50
23 CALL LBL 1 ..... CALL DRILLING PATTERN
24 L Z+100 M6 ..... TOOL CHANGE

```

Tapping

```

25 TOOL CALL 3 Z S500
26 CYCL DEF 2.0 TAPPING
27 CYCL DEF 2.1 SET UP -2
28 CYCL DEF 2.2 DEPTH -15
29 CYCL DEF 2.3 DWELL 0
30 CYCL DEF 2.4 F500
31 CALL LBL 1 ..... CALL DRILLING PATTERN

```

Retract tool, end

```

32 L Z+100 M2

```

Solution:

Linear hole pattern

SPGM 1, Drilling pattern

```
33 LBL 1 ..... DRILLING PATTERN
34 L X+5 Y+20 R0 F9999 M3 ..... PRE-POSITION
35 L Z+2 R0 M8

36 LBL 2
37 L IX+10 M99 ..... CROSS OVER

38 LBL 3
39 L IY+8 M99 ..... Y-STEP UPWARDS
40 CALL LBL 3 REP 7/7 ..... COLUMN UPWARDS

41 L IX+10 M99 ..... CROSS OVER



42 LBL 4
43 L IY-8 M99 ..... Y-STEP DOWNWARDS
44 CALL LBL 4 REP 7/7 ..... COLUMN DOWNWARDS

45 CALL LBL 2 REP 3/3 ..... REMAINING COLUMNS

46 LBL 0
47 END PGM 7109 MM
```

Task: **Linear offset hole pattern, including tapping**

Program(s): _____




Procedure: • Move back and forth along rows (meandering)

Tools: • NC-center drill
• Twist drill
• Tap

Program layout:

Linear offset hole pattern, including tapping

Meandering movement:



Left end of short row = Pre-position for next row

First pre-position lies under the first long row

Preparation

BLK FORM
TOOL DEF
TOOL 1...

Workpiece blank
Tool list
First tool call

L Z... M6

Tool change

Process Center

CYCL DEF, DEPTH = PECKG

Pecking cycle

CALL LBL 1

Call drilling pattern

STOP M6

Tool change

Pecking

TOOL 2...
CYCL DEF, DEPTH ≠ PECKG

Pecking cycle

CALL LBL 1

Call drilling pattern

STOP M6

Tool change

Tapping

TOOL 3...
CYCL DEF

Tapping cycle

CALL LBL 1

Call drilling pattern

Process end

STOP M2



HEIDENHAIN

Advanced course TNC 4xx

C05







7426/2

Program layout:

Linear offset hole pattern, including tapping


**Drilling pattern,
SPGM 1**

<i>LBL 1</i>		
<i>L X... Y...</i> <i>L Z...</i>		Absolute pre-position Setup clearance
<i>LBL 2</i>		Label
<i>L IX... IY... M99</i>		Define increments Cross over and drill
<i>LBL 3</i>		Label
<i>L IX... M99</i>		Move right
<i>CALL LBL 3 REP...</i>		
<i>L IX... IY... M99</i>		Cross over and drill
<i>LBL 4</i>		Label
<i>L IX... M99</i>		Move left
<i>CALL LBL 4 REP...</i>		
<i>CALL LBL 2 REP...</i>		Remaining rows
<i>L Z...</i> <i>L X... Y... ..</i>		Tool change position
<i>LBL 0</i>		



Solution:

Linear offset hole pattern, including tapping



Main program

```

0 BEGIN PGM 7426 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5 ..... NC-CENTER DRILL
4 TOOL DEF 2 L+0 R+2,5 ..... TWIST DRILL
5 TOOL DEF 3 L+0 R+3 ..... TAP

```

Center

```

6 TOOL CALL 1 Z S800 ..... NC-CENTER DRILL
7 L Z+100 R0 F9999 M6 ..... TOOL CHANGE
8 CYCL DEF 1.0 PECKING
9 CYCL DEF 1.1 SET UP -2
10 CYCL DEF 1.2 DEPTH -2
11 CYCL DEF 1.3 PECKG -2
12 CYCL DEF 1.4 DWELL 0
13 CYCL DEF 1.5 F200
14 CALL LBL 1 ..... CALL DRILLING PATTERN
15 STOP M6 ..... TOOL CHANGE

```

Pecking

```

16 TOOL CALL 2 Z S500 ..... TWIST DRILL
17 CYCL DEF 1.0 PECKING
18 CYCL DEF 1.1 SET UP -2
19 CYCL DEF 1.2 DEPTH -20
20 CYCL DEF 1.3 PECKG -10
21 CYCL DEF 1.4 DWELL 0
22 CYCL DEF 1.5 F100
23 CALL LBL 1 ..... CALL DRILLING PATTERN
24 STOP M6 ..... TOOL CHANGE

```

Tapping

```

25 TOOL CALL 3 Z S500 ..... TAP
26 CYCL DEF 2.0 TAPPING
27 CYCL DEF 2.1 SET UP -2
28 CYCL DEF 2.2 DEPTH -15
29 CYCL DEF 2.3 DWELL 0
30 CYCL DEF 2.4 F500
31 CALL LBL 1 ..... CALL DRILLING PATTERN

```

Retract tool, end

```

32 STOP M2

```

Solution:

Linear offset hole pattern, including tapping

SPGM 1, Drilling pattern

```
33 LBL 1
34 L X+25 Y+0 R0 F9999 M3 ..... PRE-POSITION
35 L Z+2 R0 M8

36 LBL 2 ..... CROSS OVER AND DRILL
37 L IX-5 IY+10 R0 M99

38 LBL 3 ..... RIGHT ALONG ROW
39 L IX+10 R0 M99
40 CALL LBL 3 REP 6/6

41 L IX-5 IY+10 R0 M99 ..... CROSS OVER

42 LBL 4 ..... LEFT ALONG ROW
43 L IX-10 R0 M99
44 CALL LBL 4 REP 5/5



45 CALL LBL 2 REP 3/3 ..... REMAINING ROWS

46 L Z+100 R0 M9 ..... TOOL CHANGE POSITION
47 L X-20 Y-20 R0
48 LBL 0
49 END PGM 7426 MM
```



Task: Four double bolt hole circles, with 3 tools, 3 cycles, including tapping

Program(s): _____



- Tools:**
- NC-center drill
 - Twist drill
 - Tap

BLK-FORM

Center

```
TOOL DEF / TOOL CALL 1 ...
CYCL DEF ...
CALL LBL 1
```

Pecking

```
TOOL DEF / TOOL CALL 2 ...
CYCL DEF ...
CALL LBL 1
```

Tapping

```
TOOL DEF / TOOL CALL 3 ...
CYCL DEF ...
CALL LBL 1
```

Retract tool, end

```
L Z100 M2
```

SPGM 1,
Circle center and
SPGM call

```
LBL 1
CC X... Y...
CALL LBL 2
CC X... Y...
CALL LBL 2
CC X... Y...
CALL LBL 2
LBL 0
```

SPGM 2,
Tool movements
and holes

```
LBL 2
LP PR... PA... M3
L Z2 M99
LBL 3
•
•
LBL 4
•
•
LBL 0
```

Inner bolt hole circle

Outer bolt hole circle

Program layout: **Four double bolt hole circles,
with 3 tools, 3 cycles,
including tapping**

Preparation

```
BLK FORM
TOOL DEF
TOOL 1...

L Z... M6
```

Workpiece blank
Tool list
First tool call

Tool change

**Process
Center**

```
CYCL DEF, DEPTH = PECKG
CALL LBL 1
L Z... M6
```

Pecking cycle

Call drilling pattern

Tool change

Pecking

```
TOOL 2...
CYCL DEF, DEPTH ≠ PECKG
CALL LBL 1
L Z... M6
```

Pecking cycle

Call drilling pattern

Tool change

Tapping

```
TOOL 3...
CYCL DEF
CALL LBL 1
```

Tapping cycle

Call drilling pattern

Retract tool, end

```
L Z... M2
```

Program layout:

**Four double bolt hole circles,
with 3 tools, 3 cycles,
including tapping**

**Center of circles,
SPGM 1**

<i>LBL 1</i>	
<i>CC X... Y...</i>	
<i>CALL LBL 2</i>	
<i>CC X... Y...</i>	
<i>CALL LBL 2</i>	
<i>...</i>	
<i>LBL 0</i>	

Center lower left

Call bolt hole circle

Remaining centers

Call remaining bolt
hole circles

End SPGM 1

**Drilling pattern,
SPGM 2
Bolt hole circle**

<i>LBL 2</i>	
<i>LP PR... PA... R0 F9999 M13 L Z... M99</i>	
<i>LBL 3</i>	
<i>LP PR... IPA... M99</i>	
<i>CALL LBL 3 REP...</i>	
<i>LP PR... M99</i>	
<i>LBL 4</i>	
<i>LP PR... IPA... M99</i>	
<i>CALL LBL 4 REP...</i>	
<i>LBL 0</i>	

Drill positions

Inner circle
Setup clearance
and first hole
Label

Remaining
drill positions

Outer circle,
first hole


Label

Remaining
drill positions

End SPGM 2

Solution:

**Four double bolt hole circles,
with 3 tools, 3 cycles,
including tapping**



Main program

```

0 BEGIN PGM 7139 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5 ..... NC-CENTER DRILL
4 TOOL DEF 2 L+0 R+2,5 ..... TWIST DRILL
5 TOOL DEF 3 L+0 R+3 ..... TAP
6 L Z+100 R0 F9999 M6 ..... TOOL CHANGE

```

Center

```

7 TOOL CALL 1 Z S2500 ..... NC-CENTER DRILL
8 CYCL DEF 1.0 PECKING
9 CYCL DEF 1.1 SET UP -2
10 CYCL DEF 1.2 DEPTH -1,5
11 CYCL DEF 1.3 PECKG -1,5
12 CYCL DEF 1.4 DWELL 0
13 CYCL DEF 1.5 F100
14 CALL LBL 1 ..... CALL DRILLING PATTERN
15 L Z+100 M6 ..... TOOL CHANGE

```

Pecking

```

16 TOOL CALL 2 Z S500
17 CYCL DEF 1.0 PECKING
18 CYCL DEF 1.1 SET UP -2
19 CYCL DEF 1.2 DEPTH -22
20 CYCL DEF 1.3 PECKG -10
21 CYCL DEF 1.4 DWELL 0
22 CYCL DEF 1.5 F100
23 CALL LBL 1 ..... CALL DRILLING PATTERN
24 L Z+100 M6 ..... TOOL CHANGE

```

Tapping

```

25 TOOL CALL 3 Z S250
26 CYCL DEF 2.0 TAPPING
27 CYCL DEF 2.1 SET UP -2
28 CYCL DEF 2.2 DEPTH -18
29 CYCL DEF 2.3 DWELL 0
30 CYCL DEF 2.4 F250
31 CALL LBL 1 ..... CALL DRILLING PATTERN

```

Retract tool, end

```

32 L Z+100 M2

```


Solution:

**Four double bolt hole circles,
with 3 tools, 3 cycles,
including tapping**

**SPGM 1, Center of
circles**

33 LBL 1
34 CC X+25 Y+25 CENTER LOWER LEFT
35 CALL LBL 2 DRILL POSITIONS
36 CC X+75 Y+25 CENTER LOWER RIGHT
37 CALL LBL 2 DRILL POSITIONS
38 CC X+75 Y+75 CENTER UPPER RIGHT
39 CALL LBL 2 DRILL POSITIONS
40 CC X+25 Y+75 CENTER UPPER LEFT
41 CALL LBL 2 DRILL POSITIONS
42 LBL 0

SPGM 1, end

**SPGM 2, Bolt hole
circle itself**

43 LBL 2 DRILL POSITIONS
44 LP PR+10 PA+0 M13 INNER CIRCLE
45 L Z+2 M99 1ST HOLE

46 LBL 3 REMAINING HOLES
47 LP IPA+90 M99
48 CALL LBL 3 REP 2/2


49 LP PR+18 PA+180 M99 OUTER CIRCLE

50 LBL 4 REMAINING HOLES
51 LP IPA+45 M99
52 CALL LBL 4 REP 6/6

SPGM 2, end

53 LBL 0
54 END PGM 7139 MM





- Procedure:**
- Center holes, drill, tap
 - Mill large hole

- Tools:**
- NC-center drill
 - Twist drill
 - Tap
 - Mill R30

Program layout: **Large drilled plate**

Preparation

```
BLK FORM
TOOL 1.../TOOL 2...
```

Workpiece blank
Tool list if necessary

**Process
Center**

```
TOOL 1... (R 2,0)
CYCL DEF,DEPTH = PECKG
CALL LBL 1
... M6
```

Pecking cycle
Call drilling pattern
Tool change

Pecking

```
TOOL 2... (R 1,0)
CYCL DEF
CALL LBL 1
... M6
```

Pecking cycle
Call drilling pattern
Tool change

Tapping

```
TOOL 3... (R 1,5)
CYCL DEF
CALL LBL 1
... M6
```

Tapping cycle
Call drilling pattern
Tool change

Mill large holes

```
TOOL 4... (R 30)
CYCL DEF
L X... Y...
L Z... M99
```


Circular pocket-cycle
Starting position and
cycle call

Retract tool, end

```
L Z... M2
```


**Drilling pattern,
SPGM 1**

<i>LBL 1</i>	
<i>CC...</i> <i>LP...</i>	Center Absolute pre-position Plane
<i>LZ...</i>	Setup clearance in Z
<i>LBL 2</i>	
<i>LP IPA M99</i>	Angular step (incred.) Drill with M99
<i>LBL 3</i>	
<i>LP IPR M99</i>	Radial step inwards
<i>CALL LBL 3 REP . . .</i>	Further radial steps
<i>LP IPA M99</i>	Angular step
<i>LBL 4</i>	
<i>LP IPR M99</i>	Radial step outwards
<i>CALL LBL 4 REP . . .</i>	Further radial steps
<i>CALL LBL 2 REP . . .</i>	Further double rows
<i>LP IPA</i>	To intermediate pos.
<i>CALL LBL 2 REP . . .</i>	Remaining hole gr.
<i>LZ M99</i> <i>LX . . . Y</i>	Tool change position
<i>LBL 0</i>	



Solution:

Large drilled plate



Main program

```

0 BEGIN PGM 7411 MM
1 BLK FORM 0.1 Z X-125 Y-130 Z-22
2 BLK FORM 0.2 X+125 Y+130 Z+0
3 TOOL DEF 1 L+0 R+2,0 ..... CENTER DRILL
4 TOOL DEF 2 L+0 R+1 ..... TWIST DRILL
5 TOOL DEF 3 L+0 R+1,5 ..... TAP
6 TOOL DEF 4 L+0 R+30 ..... MILL

```

Center

```

7 TOOL CALL 1 Z S3000
8 CYCL DEF 1.0 PECKING
9 CYCL DEF 1.1 SET UP -2
10 CYCL DEF 1.2 DEPTH -2
11 CYCL DEF 1.3 PECKG -2
12 CYCL DEF 1.4 DWELL 0
13 CYCL DEF 1.5 F100
14 CALL LBL 1
15 L Z+20 R0 F9999 M6

```

Pecking

```

16 TOOL CALL 2 Z S3000
17 CYCL DEF 1.0 PECKING
18 CYCL DEF 1.1 SET UP -2
19 CYCL DEF 1.2 DEPTH -20
20 CYCL DEF 1.3 PECKG -20
21 CYCL DEF 1.4 DWELL 0
22 CYCL DEF 1.5 F200
23 CALL LBL 1
24 L Z+20 R0 F9999 M6

```



HEIDENHAIN

Advanced course TNC 4xx

C04



7411/5

Solution: Large drilled plate

Tapping

```

25 TOOL CALL 3 Z S500
26 CYCL DEF 2.0 TAPPING
27 CYCL DEF 2.1 SET UP -2
28 CYCL DEF 2.2 DEPTH -17
29 CYCL DEF 2.3 DWELL 0
30 CYCL DEF 2.4 F250
31 CALL LBL 1
32 L Z+20 R0 F9999 M6
  
```

Circular pocket

```

33 TOOL CALL 4 Z S400
34 CYCL DEF 5.0 CIRCULAR POCKET
35 CYCL DEF 5.1 SET UP -2
36 CYCL DEF 5.2 DEPTH -15
37 CYCL DEF 5.3 PECKG -5 F100
38 CYCL DEF 5.4 RADIUS 75
39 CYCL DEF 5.5 F500 DR-
40 L X+0 Y+0 R0 M3
41 L Z+2 M99
  
```

Retract tool, end

```

42 L Z+20 R0 F9999 M2
  
```

SPGM 1, Drilling pattern

```

43 LBL 1
44 CC X+0 Y+0 ..... CENTER
45 LP PR+120 PA-6,666 R0 F9999 M3 ..... PRE-POSITION BY 1ST HOLE
46 L Z+2 R0 M8

47 LBL 2
48 LP IPA+3,333 M99 ..... ANGULAR STEP


49 LBL 3
50 LP IPR-8 M99
51 CALL LBL 3 REP 2/2 ..... ROW RADIALLY INWARDS

52 LP IPA+3,333 M99 ..... ANGULAR STEP

53 LBL 4
54 LP IPR+8 M99
55 CALL LBL 4 REP 2/2 ..... ROW RADIALLY OUTWARDS


56 CALL LBL 2 REP 1/1 ..... FURTHER DOUBLE ROWS

57 LP IPA+4,666 ..... SPACE
58 CALL LBL 2 REP 19/19 ..... REPEAT FURTHER GROUPS OF 16
59 LBL 0
60 END PGM 7411 MM
  
```



Task: **Loading frame for plate
with M97 and M98**

Program(s): _____



Procedure:

- Rough out inside
- Travel one mill radius farther at the corners of the loading surface
- Small contour step on the side protrusions

Tools:

- Roughing-finishing mill R5



HEIDENHAIN

Advanced course TNC 4xx

C04



72810/1

Program layout:

Loading frame for plate with M97 and M98

Preparation

```
BLK FORM  
TOOL DEF  
TOOL 1...  
  
L Z... .. M6
```

Workpiece blank
Tool list
Call tool data
Tool change

Process

Rough out inside

```
CYCL DEF  
L X... Y...  
L Z... .. M99
```

Pocket milling-cycle
Starting position and
cycle call

Loading surface

```
L X... Y... R0  
L Z...  
APPR... X... Y...  
  
L X... M98  
  
L X... Y...  
L Y... M97  
  
L X...  
L Y...  
L X... M97  
  
L Y... M98  
  
L X...  
L X... M98  
  
L X...  
DEP... X... Y...
```

Pre-position
Depth
Contour approach tang. ,
lower center
In corner one tool
radius farther

Small contour step

Protrusion
Protrusion
Reverse contour step

Next corner

Next corner


End of contour
Depart contour tang.

Retract tool, end

```
L Z... .. M2
```

Tool change position





```

0 BEGIN PGM 72810 MM
1 ..... LOADING FRAME WITH APPR/DEP
2 BLK FORM 0.1 Z X+0 Y+0 Z-25
3 BLK FORM 0.2 X+110 Y+110 Z+0
4 TOOL DEF 1 L+0 R+5
5 TOOL CALL 1 Z S2500
6 L Z+100 R0 F9999 M6
7 CYCL DEF 4.0 POCKET MILLING
8 CYCL DEF 4.1 SET UP -2
9 CYCL DEF 4.2 DEPTH -10
10 CYCL DEF 4.3 PECKG -10 F150
11 CYCL DEF 4.4 X+82
12 CYCL DEF 4.5 Y+82
13 CYCL DEF 4.6 F1000 DR-
14 L X+55 Y+55 R0 M3
15 L Z+2 R0 M99

16 L X+30 Y+30
17 L Z-5 R0 F100
18 APPR LCT X+30 Y+10 R5 RR F250
19 L X+10 M98
20 L X+10 Y+15
21 L Y+50 M97
22 L X+8
23 L Y+90
24 L X+10 M97
25 L Y+100 M98
26 L X+15
27 L X+100 M98
28 L Y+95
29 L Y+10 M98
30 L X+95
31 L X+30
32 DEP LCT X+30 Y+30 R5

33 L Z+100 R0 F9999 M2
34 END PGM 72810 MM
    
```



Roughing out

Milling




Task: **Double L-form with SL II-cycles**

Program(s): _____



SL-cycles, program flowchart



Program layout: **Double L-form with SL II-cycles**

Preparation

List part contours

BLK FORM
CYCL 14.0 CONTOUR GEOMETRY
CYCL 14.1 CONTOUR LABEL 1
CYCL 20 CONTOUR DATA

General data for production (DEF-cycles)

Pilot drilling

TOOL 1 . . .
CYCL 21 PILOT DRILLING
 ⋮
CALL LBL 10

CALL-cycles

Roughing out

TOOL 2 . . .
CYCL 22 ROUGH OUT
 ⋮
CALL LBL 10

Floor finishing

TOOL 3 . . .
CYCL 23 FLOOR FINISHING
 ⋮
CALL LBL 10

Side finishing

CYCL 24 SIDE FINISHING
 ⋮
CALL LBL 10

Retract tool, end

L Z . . . M2

SPGM 10

LBL 10

CYCL 7.0 DATUM SHIFT
CYCL CALL

CYCL 7.0 DATUM SHIFT

CYCL 10.0 ROTATION
CYCL 10.1 ROT . . .
CYCL CALL

CYCL 10.0 ROTATION
CYCL 10.1 ROT+0


LBL 0

Primary subprogram

**Part contours,
 SPGM 1 . . . SPGM n**

LBL 1 . . . LBL 0
 ⋮
LBL n . . . LBL 0





Main program

```

0 BEGIN PGM 7172 MM
1 ..... DOUBLE L-FORM WITH
2 ..... 20-SERIES CYCLES
3 BLK FORM 0.1 Z X+0 Y+0 Z-22
4 BLK FORM 0.2 X+100 Y+100 Z+0
5 CYCL DEF 14.0 CONTOUR GEOMETRY
6 CYCL DEF 14.1 CONTOUR LABEL 1
7 CYCL DEF 20.0 CONTOUR DATA
  Q1=-20 ..... MILLING DEPTH
  Q2=1 ..... TOOL PATH OVERLAP
  Q3=+1 ..... ALLOWANCE FOR SIDE
  Q4=+1 ..... ALLOWANCE FOR FLOOR
  Q5=+0 ..... WORKPIECE SURFACE COORD.
  Q6=-2 ..... SET UP CLEARANCE
  Q7=+20 ..... CLEARANCE HEIGHT
  Q8=+0 ..... ROUNDING RADIUS
  Q9=-1 ..... DIRECTION OF ROTATION
8 TOOL DEF 1 L+0 R+3
9 TOOL CALL 1 Z S2000
10 L Z+100 R0 F9999
11 L X-20 Y-20 R0 M3
12 CYCL DEF 21.0 PILOT DRILLING
  Q10=-5 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q13=2 ..... ROUGH OUT TOOL
13 L Z+20 R0 F9999 M3
14 CALL LBL 10

15 TOOL DEF 2 L+0 R+4
16 TOOL CALL 2 Z S3150
17 STOP
18 CYCL DEF 22.0 ROUGH OUT
  Q10=-10 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=200 ..... FEED RATE FOR MILLING
19 L Z+2 R0 F MAX M3
20 CALL LBL 10
    
```



Solution:

Double L-form with SL II-cycles

```
21 L Z+20 R0 F MAX M6

22 TOOL DEF 3 L+0 R+3
23 TOOL CALL 3 Z S3150
24 CYCL DEF 23.0 FLOOR FINISHING
    Q11=100 ..... FEED RATE FOR PLUNGING
    Q12=500 ..... FEED RATE FOR MILLING
25 CALL LBL 10
26 STOP

27 CYCL DEF 24.0 SIDE FINISHING
    Q9=+1 ..... DIRECTION OF ROTATION
    Q10=-10 ..... PLUNGING DEPTH
    Q11=100 ..... FEED RATE FOR PLUNGING
    Q12=500 ..... FEED RATE FOR MILLING
    Q14=+0 ..... ALLOWANCE FOR SIDE
28 L Z+2 R0 F MAX M3
29 CALL LBL 10

Retract tool, end    30 L Z+100 R0 F MAX M2
```

SPGM 10

```
31 LBL 10
32 CYCL DEF 7.0 DATUM SHIFT
33 CYCL DEF 7.1 X+20
34 CYCL DEF 7.2 Y+12
35 CYCL CALL M3

36 CYCL DEF 7.0 DATUM SHIFT
37 CYCL DEF 7.1 Y+50
38 CYCL DEF 7.2 X+30
39 CYCL DEF 10.0 ROTATION
40 CYCL DEF 10.1 ROT+35
41 CYCL CALL
```

UP10, end

```
42 CYCL DEF 10.0 ROTATION
43 CYCL DEF 10.1 ROT+0
44 LBL 0
```

SPGM 1

```
45 LBL 1
46 L X+10 Y+0 RR
47 L X+0
48 RND R6
49 L Y+30
50 RND R6
51 L X+15
52 RND R6
53 L Y+15
54 L X+60
55 RND R6
56 L Y+0
57 RND R6
58 L X+10
59 LBL 0
60 END PGM 7172 MM
```

SPGM 1, end



HEIDENHAIN

Advanced course TNC 4xx



C04



7172/4

Task: **Mixed elements (internal) with with Q-parameters**

Program(s): _____



Program layout:

Use of SL II-cycles for in depth step contour

Preparation

Roughing-out

```
BLK FORM
TOOL 1... R3,5
CYCL 20.0 CONTOUR DATA
... = 0,4
... = 0
... = -2...
CYCL 22.0 ROUGH OUT...
```

Oversize for side
Surface
Setup clearance

```
CALL LBL 10
```

Finishing

```
TOOL 2... R2
CYCL 24.0 SIDE FINISHING
```

Oversize for side

```
... = 0
```

```
CALL LBL 10
```

“Clean out” corners

```
TOOL 3... R1
```

```
CALL LBL 10
```

Retract tool, end

```
L Z... M2
```

SPGM 10

```
LBL 10
```

First level

```
L Z20... M3
CYCL 14 CONTOUR LABEL 1/2
Q5 = 0
L X... Y... M99
```

Surface

Second level

```
CYCL 14 CONTOUR LABEL 1/2/3
Q5 = -8
CYCL CALL
```

Surface

Third level

```
CYCL 14 CONTOUR LABEL 2/3
Q5 = -16
Q6 = -10
CYCL CALL
```

Surface
Setup clear. increased

```
LBL 0
```

Part contours,
SPGM 1, SPGM 2,
SPGM 3

```
LBL 1... LBL 0
```

Rectangular pocket

```
LBL 2... LBL 0
```

Circular pocket

```
LBL 3... LBL 0
```

Oval island



HEIDENHAIN

Advanced course TNC 4xx


C01



761541/2

Solution:

Mixed elements (internal) with with Q-parameters



Main program

```

0 BEGIN PGM 761541 MM
1 ..... CARBURETOR CHAMBER (INSIDE)
2 BLK FORM 0.1 Z X+0 Y+15 Z-25
3 BLK FORM 0.2 X+100 Y+95 Z+0
4 TOOL DEF 1 L+0 R+3,5
5 TOOL CALL 1 Z S3000
6 CYCL DEF 20.0 CONTOUR DATA
  Q1=-8 ..... MILLING DEPTH
  Q2=1 ..... TOOL PATH OVERLAP
  Q3=+0,4 ..... ALLOWANCE FOR SIDE
  Q4=+0 ..... ALLOWANCE FOR FLOOR
  Q5=+0 ..... WORKPIECE SURFACE COORD.
  Q6=-2 ..... SET UP CLEARANCE
  Q7=+10 ..... CLEARANCE HEIGHT
  Q8=+0 ..... ROUNDING RADIUS
  Q9=-1 ..... DIRECTION OF ROTATION
7 CYCL DEF 22.0 ROUGH OUT
  Q10=-4 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=500 ..... FEED RATE FOR MILLING
8 CALL LBL 10
9 STOP M6

10 TOOL DEF 2 L+0 R+2
11 TOOL CALL 2 Z S3000
12 CYCL DEF 24.0 SIDE FINISHING
  Q9=+1 ..... DIRECTION OF ROTATION
  Q10=-4 ..... PLUNGING DEPTH
  Q11=500 ..... FEED RATE FOR PLUNGING
  Q12=1000 ..... FEED RATE FOR MILLING
  Q14=+0 ..... ALLOWANCE FOR SIDE
13 CALL LBL 10
14 STOP M6

15 TOOL DEF 3 L+0 R+1
16 TOOL CALL 3 Z S4000
17 CALL LBL 10

Retract tool, end
18 L Z+50 R0 F9998 M2
  
```



HEIDENHAIN

Advanced course TNC 4xx

C05



761541/3

Solution: **Mixed elements (internal) with
with Q-parameters**

SPGM 10

19 LBL 10
20 L Z+20 R0 F9999 M3
21 CYCL DEF 14.0 CONTOUR GEOMETRY
22 CYCL DEF 14.1 CONTOUR LABEL 1/2
23 FN 0: Q5 = +0 SURFACE
24 L X+10 Y+25 R0 F9998 M99
25 CYCL DEF 14.0 CONTOUR GEOMETRY
26 CYCL DEF 14.1 CONTOUR LABEL 1/2/3
27 FN 0: Q5 = -8 SURFACE
28 CYCL CALL

29 CYCL DEF 14.0 CONTOUR GEOMETRY
30 CYCL DEF 14.1 CONTOUR LABEL 2/3
31 FN 0: Q5 = -16 SURFACE
32 FN 0: Q6 = -10 CLEARANCE
33 CYCL CALL
34 LBL 0

**SPGM 1,
rectangular pocket**

35 LBL 1
36 L X+5 Y+20 RR
37 L Y+75
38 L X+60
39 L Y+20
40 L X+5
41 LBL 0

**SPGM 2,
circular pocket**

42 LBL 2
43 CC X+60 Y+55
44 LP PR+35 PA+0 RR
45 CP DR-
46 LBL 0


SPGM 3, island

47 LBL 3
48 L X+55 Y+30 RL
49 L X+25
50 CT Y+50
51 L X+55
52 CT Y+30
53 LBL 0
54 END PGM 761541 MM




Task: **Mixed elements (internal and external) with Q-parameters**

Program(s): _____



Solution:

Mixed elements (internal and external) with Q-parameters



Main program

```
0 BEGIN PGM 761591 MM
1 ..... CARBURETOR CHAMBER (INSIDE
2 ..... AND OUTSIDE), TNC 4XX
3 BLK FORM 0.1 Z X+0 Y+15 Z-25
4 BLK FORM 0.2 X+100 Y+95 Z+0
5 TOOL DEF 1 L+0 R+3,5
6 TOOL CALL 1 Z S3000
7 L Z+20 R0 F9998 M3
8 CYCL DEF 20.0 CONTOUR DATA
  Q1=-24 ..... MILLING DEPTH
  Q2=1 ..... TOOL PATH OVERLAP
  Q3=+0 ..... ALLOWANCE FOR SIDE
  Q4=+0 ..... ALLOWANCE FOR FLOOR
  Q5=+0 ..... WORKPIECE SURFACE COORD.
Q6=-2 ..... SET-UP CLEARANCE
  Q7=+10 ..... CLEARANCE HEIGHT
  Q8=+0 ..... ROUNDING RADIUS
  Q9=-1 ..... DIRECTION OF ROTATION
9 CYCL DEF 22.0 ROUGH OUT
  Q10=-4 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=500 ..... FEED RATE FOR MILLING
10 CYCL DEF 14.0 CONTOUR GEOMETRY
11 CYCL DEF 14.1 CONTOUR LABEL 4 /5 /6
12 CYCL CALL M3
13 CALL LBL 10
14 STOP M6
```

Roughing out

Finishing

```
15 TOOL DEF 2 L+0 R+2
16 TOOL CALL 2 Z S3000
17 CYCL DEF 24.0 SIDE FINISHING
  Q9=+1 ..... DIRECTION OF ROTATION
  Q10=-4 ..... PLUNGING DEPTH
Q11=500 ..... FEED RATE FOR PLUNGING
  Q12=1000 ..... FEED RATE FOR MILLING
  Q14=+0 ..... ALLOWANCE FOR SIDE
18 CALL LBL 10
19 STOP M6
```



HEIDENHAIN

Advanced course TNC 4xx

C04



761591/2

Solution:

Mixed elements (internal and external) with Q-parameters

20 TOOL DEF 3 L+0 R+1
21 TOOL CALL 3 Z S4000
22 CALL LBL 10

23 L Z+50 R0 F9998 M2

SPGM

24 LBL 10
25 CYCL DEF 14.0 CONTOUR GEOMETRY
26 CYCL DEF 14.1 CONTOUR LABEL 1/2
27 FN 0: Q1 = -8 DEPTH
28 FN 0: Q3 = +0,4 OVERSIZE FOR SIDE
29 FN 0: Q5 = +0 SURFACE
30 FN 0: Q6 = -2 SETUP CLEARANCE
31 CYCL CALL M3
32 CYCL DEF 14.0 CONTOUR GEOMETRY
33 CYCL DEF 14.1 CONTOUR LABEL 1/2/3
34 FN 0: Q5 = -8 SURFACE
35 CYCL CALL
36 CYCL DEF 14.0 CONTOUR GEOMETRY
37 CYCL DEF 14.1 CONTOUR LABEL 2/3
38 FN 0: Q5 = -16 SURFACE
39 FN 0: Q1 = -10 CLEARANCE
40 CYCL CALL

Internal elements

41 LBL 0
42 LBL 1
43 L X+5 Y+20 RR
44 L Y+75
45 L X+60
46 L Y+20
47 L X+5
48 LBL 0
49 LBL 2
50 CC X+60 Y+55
51 LP PR+35 PA+0 RR
52 CP DR-
53 LBL 0
54 LBL 3
55 L X+55 Y+30 RL
56 L X+25
57 CT Y+50
58 L X+55
59 CT Y+30
60 LBL 0

External elements

61 LBL 4
62 L X-12 Y+0 RR
63 L Y+105
64 L X+110
65 L Y+0
66 L X-12
67 LBL 0
68 LBL 5
69 L X+0 Y+15 RL
70 L Y+80
71 L X+55
72 L Y+15
73 L X+0
74 LBL 0
75 LBL 6
76 CC X+60 Y+55
77 LP PR+40 PA+0 RL
78 CP DR-
79 LBL 0
80 END PGM 761591 MM




HEIDENHAIN


Advanced course TNC 4xx

C01




761591/3





Solution:

FK Telephone



FK Program

```

0 BEGIN PGM 75103 MM
1 BLK FORM 0.1 Z X+0 Y+50 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S2800
5 L X+50 Y+50 R0 F9999 M13
6 L Z-5 F100
7 L X+50 Y+75 RL .....①
8 RND R5 F100

9 FC DR+ R25 CCX+50 CCY+50 F500 .....②
10 FCT DR- R14 .....③
11 FCT DR- R88 CCX+50 CCY+0 .....④
12 FCT DR- R14 .....⑤
13 FCT X+50 Y+75 DR+ R25 CCX+50 CCY+50 .....⑥
14 FSELECT 2

15 RND R5 F100
16 L X+50 Y+50 R0 M9

17 L Z+10
18 END PGM 75103 MM
    
```

Converted Program

```

0 BEGIN PGM BOGEN MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3
4 TOOL CALL 1 Z S2800
5 L X+50 Y+50 R0 F9999 M13
6 L Z-5 F100
7 L X+50 Y+75 RL
8 RND R5 F100

9 CC X+50 Y+50
10 C X+26,805 Y+59,3269 DR+
11 CC X+13,8158 Y+64,55
12 C X+6,9701 Y+76,7622 DR-
13 CC X+50 Y+0
14 C X+93,0299 Y+76,7622 DR-
15 CC X+86,1842 Y+64,55
16 C X+73,195 Y+59,3269 DR-
17 CC X+50 Y+50
18 C X+50 Y+75 DR+

19 RND R5 F100
20 L X+50 Y+50 R0 M9

21 L Z+10
22 END PGM BOGEN MM
    
```





HEIDENHAIN

Advanced course TNC 4xx

C01




75103/2



Solution:

FK Dumbbell



```
0 BEGIN PGM 75116 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+4
4 TOOL CALL 1 Z S4000
5 L Z+2 R0 F MAX M3
6 L X+40 Y+30 R0 F2000
7 L Z-10 F100
8 L X+10 Y+30 RR F200
9 FPOL X+40 Y+30
10 FC DR- R30 CCX+40 CCY+30 F500
11 FL AN+60 PDX+40 PDY+30 D10
12 FSELECT 3
13 FC DR- R20 CCPR+55 CCPA+60
14 FSELECT 2
15 FL AN-120 PDX+40 PDY+30 D10
16 FSELECT 3
17 FC X+10 Y+30 DR- R30 CCX+40 CCY+30
18 FSELECT 2
19 L X+40 Y+30 R0
20 L Z+50 R0 F MAX M2
21 END PGM 75116 MM
```





HEIDENHAIN

Advanced course TNC 4xx

C01




75116/2



Solution:

FK Hammer



```
0 BEGIN PGM 75119 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+85 Z+0
3 TOOL DEF 1 L+0 R+5
4 TOOL CALL 1 Z S4000
5 L Z+50 R0 F MAX M6
6 L X+80 Y+70 F MAX M3
7 L Z-10 F100
8 CC X+30 Y+40
9 LP PR+37,5 PA+30 RL F500
10 FL LEN10 AN-60
11 FL AN-150
12 FCT DR+ R12,5
13 FLT AN-100
14 FCT DR- R11 CCX+30 CCY+15
15 FLT PDX+15 PDY+65 D11
16 FSELECT 1
17 FCT DR- R11 CCX+15 CCY+65
18 FLT AN-55
19 FCT DR+ R15
20 FPOL X+30 Y+40
21 FLT AN+30 PDX+30 PDY+40 D10
22 FSELECT 2
23 FL PR+37,5 PA+30 IAN+90
24 L X+80 Y+70 R0 F500
25 L Z+50 R0 F MAX M2
26 END PGM 75119 MM
```



HEIDENHAIN

Advanced course TNC 4xx

C01



75119/2

Program layout: **FK Hammer, island with 20-series cycles, with one tool**


Preparation	<i>BLK FORM</i>	
	<i>TOOL 1...</i>	
Listing	<i>CYCL 14 CONTOUR LABEL 1/2</i>	Part contours
	<i>CYCL 20 CONTOUR DATA</i>	Contour data
	<i>CYCL 22 ROUGH OUT</i>	Roughing-out
Pre-position	<i>L X... Y... F9999 M3</i>	Plane (due to clamping devices)
	<i>CYCL CALL</i>	Call
	<i>CYCL 23 FLOOR FINISHING</i>	Floor finishing
	<i>CYCL CALL</i>	Call
	<i>CYCL 24 SIDE FINISHING</i>	Finishing
	<i>CYCL CALL</i>	
Retract tool, end	<i>L Z100 M2</i>	

SPGM, part contours	<i>LBL 1</i>	Island Contour only! Without approach and departure
	<i>L X... Y... RL</i>	
	• • •	
	<i>LBL 0</i>	
	<i>LBL 2</i>	Pocket Contour only!
	<i>L X... Y... RR</i>	
	• • •	
	<i>LBL 0</i>	



Solution:

FK Hammer, island with 20-series cycles



Main program

```

0 BEGIN PGM 7173 MM
1 ..... HAMMER HEAD WITH 20-SERIES
2 ..... CYCLES, BO6
3 BLK FORM 0.1 Z X+0 Y+0 Z-20
4 BLK FORM 0.2 X+70 Y+80 Z+0
5 TOOL DEF 5 L+0 R+5
6 TOOL CALL 5 Z S2500
7 CYCL DEF 14.0 CONTOUR GEOMETRY
8 CYCL DEF 14.1 CONTOUR LABEL 1/2
9 CYCL DEF 20.0 CONTOUR DATA
  Q1=-10 ..... MILLING DEPTH
  Q2=1 ..... TOOL PATH OVERLAP
  Q3=+1 ..... ALLOWANCE FOR SIDE
  Q4=+1 ..... ALLOWANCE FOR FLOOR
  Q5=+0 ..... WORKPIECE SURFACE COORD.
  Q6=-2 ..... SET UP CLEARANCE
  Q7=+100 ..... CLEARANCE HEIGHT
  Q8=+0 ..... ROUNDING RADIUS
  Q9=-1 ..... DIRECTION OF ROTATION
10 CYCL DEF 22.0 ROUGH OUT
  Q10=-5 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=500 ..... FEED RATE FOR MILLING
11 L X+80 Y+70 R0 F9999 M3
12 CYCL CALL

13 CYCL DEF 23.0 FLOOR FINISHING
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=500 ..... FEED RATE FOR MILLING
14 CYCL CALL

15 CYCL DEF 24.0 SIDE FINISHING
  Q9=-1 ..... DIRECTION OF ROTATION
  Q10=-5 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=500 ..... FEED RATE FOR MILLING
  Q14=+0 ..... ALLOWANCE FOR SIDE
16 CYCL CALL

Retract tool, end
17 L Z+100 R0 F MAX M2
  
```



HEIDENHAIN

Advanced course TNC 4xx

C01



7173/3

Solution:

FK Hammer, island with 20-series cycles

SPGM

```
18 LBL 1
19 L X+30 Y+4 RL
20 FC DR- R11 CCX+30 CCY+15
21 FLT
22 FCT DR- R11 CCA+35 CCX+15 CCY+65
23 FLT
24 FCT DR+ R15
25 FLT AN+30 PDX+30 PDY+40 D10
26 FSELECT 2
27 FL LEN20 AN-60 PDX+30 PDY+40 D37,5
28 FL AN+210 PDX+30 PDY+40 D10
29 FCT DR+ R12,5
30 FLT AN-100
31 FCT X+30 DR- R11 CCX+30 CCY+15
32 FSELECT 2
33 LBL 0

34 LBL 2
35 L X+30 Y-15 RR
36 L X-8
37 L Y+90
38 L X+90
39 L Y-15
40 L X+30
41 LBL 0
42 END PGM 7173 MM
```




HEIDENHAIN

Advanced course TNC 4xx

C01




7173/4



Solution:


FK Mickey Mouse



```
0 BEGIN PGM 71750 MM
1 ..... MICKEY
2 BLK FORM 0.1 Z X-70 Y-90 Z-12
3 BLK FORM 0.2 X+70 Y+30 Z+0
4 TOOL DEF 1 L+0 R+5
5 TOOL CALL 1 Z S2500
6 L Z+10 R0 F9999 M3
7 L X-62 Y+0 RL
8 L Z-10 F150
9 FC DR- R22 CLSD+ CCX-40 CCY+0
10 FCT DR+ R38 CCX+0
11 FSELECT 1
12 FCT DR- R22 CCX+40 CCY+0
13 FCT DR+ R10
14 FLT AN-60 PDX+0 PDY+0 D0
15 FSELECT 3
16 FCT DR- R8
17 FCT X+0 Y-80 DR- R80 CCX+0 CCY+0
18 FLT X-40 AN+180
19 FL AN+90
20 FL AN+30
21 FCT DR+ R9 CCX-17
22 FLT X-30 PAR21 DP18
23 FSELECT 2
24 FL AN+90
25 FL PAR23 DP8
26 FSELECT 2
27 FCT DR+ R9 CCX-17
28 FCT X-62 Y+0 DR- R22 CLSD- CCX-40 CCY+0
29 FSELECT 2
30 L Z+10 F9999 M2
31 END PGM 71750 MM
```

Solution:

FK Mickey Mouse with 20-series cycles



Main program

```

0 BEGIN PGM 7175 MM
1 ..... MICKEY WITH 20-SERIES CYCLES
2 BLK FORM 0.1 Z X-70 Y-90 Z-12
3 BLK FORM 0.2 X+70 Y+30 Z+0
4 TOOL DEF 5 L+0 R+5
5 TOOL CALL 5 Z S2500
6 L Z+20 R0 F9999 M3
7 CYCL DEF 14.0 CONTOUR GEOMETRY
8 CYCL DEF 14.1 CONTOUR LABEL 1/2
9 CYCL DEF 20.0 CONTOUR DATA
  Q1=-10 ..... MILLING DEPTH
  Q2=1 ..... TOOL PATH OVERLAP
  Q3=+1 ..... ALLOWANCE FOR SIDE
  Q4=+1 ..... ALLOWANCE FOR FLOOR
  Q5=+0 ..... WORKPIECE SURFACE COORD.
  Q6=-2 ..... SET UP CLEARANCE
  Q7=+20 ..... CLEARANCE HEIGHT
  Q8=+0 ..... ROUNDING RADIUS
  Q9=-1 ..... DIRECTION OF ROTATION
10 CYCL DEF 22.0 ROUGH OUT
  Q10=-5 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=200 ..... FEED RATE FOR MILLING
11 L X-65 Y+0 M99

12 CYCL DEF 23.0 FLOOR FINISHING
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=200 ..... FEED RATE FOR MILLING
13 CYCL CALL

14 CYCL DEF 24.0 SIDE FINISHING
  Q9=-1 ..... DIRECTION OF ROTATION
  Q10=-5 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=200 ..... FEED RATE FOR MILLING
  Q14=+0 ..... ALLOWANCE FOR SIDE
15 CYCL CALL

Retract tool, end
16 L Z+100 R0 F MAX M2
  
```

Solution:

FK Mickey Mouse with 20-series cycles

SPGM

```
17 LBL 1
18 L X-62 Y+0 RL
19 FC DR- R22 CLSD+ CCX-40 CCY+0
20 FCT DR+ R38 CCX+0
21 FSELECT 1
22 FCT DR- R22 CCX+40 CCY+0
23 FCT DR+ R10
24 FLT AN-60 PDX+0 PDY+0 D0
25 FSELECT 3
26 FCT DR- R8
27 FCT X+0 Y-80 DR- R80 CCX+0 CCY+0
28 FLT X-40 AN+180
29 FL AN+90
30 FL AN+30
31 FCT DR+ R9 CCX-17
32 FLT X-30 PAR31 DP18
33 FSELECT 2
34 FL AN+90
35 FL PAR33 DP8
36 FSELECT 2
37 FCT DR+ R9 CCX-17
38 FCT DR- R22 CLSD- CCX-40 CCY+0
39 FSELECT 2
40 LBL 0

41 LBL 2
42 L X-74 Y+20 RR
43 L Y+35
44 L X+74
45 L Y-100
46 L X-74
47 L Y+20
48 LBL 0
49 END PGM 7175 MM
```




HEIDENHAIN

Advanced course TNC 4xx

C05




7175/4



Solution:

FK T-Contour with 20-series cycles



```

0 BEGIN PGM 7174 MM
1 ..... T-CONTOUR WITH 20-SERIES
2 ..... CYCLES, B06
3 BLK FORM 0.1 Z X-70 Y-100 Z-20
4 BLK FORM 0.2 X+70 Y+5 Z+0
5 TOOL DEF 1 L+0 R+6
6 TOOL CALL 1 Z S2000
7 L Z+20 R0 F9999 M3
8 CYCL DEF 14.0 CONTOUR GEOMETRY
9 CYCL DEF 14.1 CONTOUR LABEL 1/2
10 CYCL DEF 20.0 CONTOUR DATA
    Q1=-10 ..... MILLING DEPTH
    Q2=1 ..... TOOL PATH OVERLAP
    Q3=+1 ..... ALLOWANCE FOR SIDE
    Q4=+1 ..... ALLOWANCE FOR FLOOR
    Q5=+0 ..... WORKPIECE SURFACE COORD.
    Q6=+2 ..... SET UP CLEARANCE
    Q7=+20 ..... CLEARANCE HEIGHT
    Q8=+0 ..... ROUNDING RADIUS
    Q9=-1 ..... DIRECTION OF ROTATION
11 CYCL DEF 22.0 ROUGH OUT
    Q10=-5 ..... PLUNGING DEPTH
    Q11=100 ..... FEED RATE FOR PLUNGING
    Q12=500 ..... FEED RATE FOR MILLING
12 L X+0 Y+20 M8
13 L Z+2 R0 F MAX M99
14 CYCL DEF 23.0 FLOOR FINISHING
    Q11=100 ..... FEED RATE FOR PLUNGING
    Q12=500 ..... FEED RATE FOR MILLING
15 CYCL CALL

16 CYCL DEF 24.0 SIDE FINISHING
    Q9=-1 ..... DIRECTION OF ROTATION
    Q10=-5 ..... PLUNGING DEPTH
    Q11=100 ..... FEED RATE FOR PLUNGING
    Q12=500 ..... FEED RATE FOR MILLING
    Q14=+0 ..... ALLOWANCE FOR SIDE
17 CYCL CALL

18 L Z+100 R0 F MAX M2
  
```



HEIDENHAIN

Advanced course TNC 4xx

C01



7174/2

Solution:

FK T-Contour with 20-series cycles

SPGM

```
19 LBL 1
20 L X+0 Y+0 RL
21 FL AN+0
22 FCT X+60 DR- R6
23 FLT AN+270
24 FL X+48 Y-16 AN+198
25 L X+8
26 RND R6
27 L Y-75
28 RND R6
29 FL AN+0
30 FCT DR+ R6
31 FLT AN+125
32 FCT DR- R5 CCX+33
33 FLT X+60 Y-64 AN+305
34 L Y-96
35 RND R7,5
36 L X-60
37 RND R7,5
38 L Y-64
39 FL AN+55
40 FCT DR- R5 CCX-33
41 FLT AN+235
42 FCT Y-75 DR+ R6
43 FLT AN+0
44 FCT X-8 DR+ R6
45 FLT AN+90
46 FCT Y-16 DR+ R6
47 FLT X-48 Y-16 AN+180
48 FL X-60 AN+162
49 FL AN+90
50 FCT X-54 Y+0 DR- R6
51 FSELECT 2
52 L X+0
53 LBL 0

54 LBL 2
55 L X+100 Y+20 RR
56 L Y-120
57 L X-100
58 L Y+20
59 L X+100
60 LBL 0
61 END PGM 7174 MM
```




HEIDENHAIN

Advanced course TNC 4xx

C05




7174/3



Solution:

FK Comb



```
0 BEGIN PGM 75108 MM
1 BLK FORM 0.1 Z X-80 Y-10 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+3,5
4 TOOL CALL 1 Z S1000
5 L X-100 Y+0 R0 F MAX M3
6 L Z-20 F MAX
7 L X-80 Y+0 RL F600
8 FPOL X+0 Y+0
9 FC DR- R80 CCX+0 CCY+0
10 FCT DR- R8 CCPR+72 CCPA+120
11 FCT DR+ R64 CCX+0 CCY+0
12 FCT DR+ R8 CCY+8
13 FSELECT 2
14 FCT DR- R48 CCX+0 CCY+0
15 FCT DR- R8 CCPR+40 CCPA+120
16 FCT DR+ R32 CCX+0 CCY+0
17 FCT DR+ R6 CCY+6
18 FSELECT 2
19 FCT X+0 Y+20 DR- R20 CCX+0 CCY+0
20 FLT X+12 Y+20
21 FL AN-60
22 FCT Y+0 DR+ R4
23 FLT AN+0
24 FCT DR+ R4
25 FLT AN+120 PAR21 DP16
26 FCT DR- R8 CCY+34
27 FLT AN-60 PAR25 DP16
28 FCT Y+0 DR+ R4
29 FLT Y+0 AN+0
30 FCT DR+ R4
31 FLT AN+120 PAR27 DP16
32 FCT DR- R8 CCY+64
33 FLT AN-60 PAR31 DP16
34 FCT X+100 Y+0 DR+ R4
35 FSELECT 2
36 FL X+100 Y-20
37 FL X-80 Y-20
38 FL X-80 Y+0
39 END PGM 75108 MM
```




HEIDENHAIN

Advanced course TNC 4xx

C05



75108/2



Main program

```

0 BEGIN PGM 7176 MM
1 ..... FINGER WITH 20-SERIES CYCLES
2 BLK FORM 0.1 Z X-85 Y-25 Z-12
3 BLK FORM 0.2 X+105 Y+75 Z+0
4 TOOL DEF 5 L+0 R+5
5 TOOL CALL 5 Z S2500
6 L Z+20 R0 F9999 M3
7 CYCL DEF 14.0 CONTOUR GEOMETRY
8 CYCL DEF 14.1 CONTOUR LABEL 1/2
9 CYCL DEF 20.0 CONTOUR DATA
  Q1=-10 ..... MILLING DEPTH
  Q2=1 ..... TOOL PATH OVERLAP
  Q3=+1 ..... ALLOWANCE FOR SIDE
  Q4=+1 ..... ALLOWANCE FOR FLOOR
  Q5=+0 ..... WORKPIECE SURFACE COORD.
  Q6=-2 ..... SET UP CLEARANCE
  Q7=+20 ..... CLEARANCE HEIGHT
  Q8=+0 ..... ROUNDING RADIUS
  Q9=-1 ..... DIRECTION OF ROTATION
10 CYCL DEF 22.0 ROUGH OUT
  Q10=-10 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=200 ..... FEED RATE FOR MILLING
11 L X-100 Y-20 M99

12 CYCL DEF 23.0 FLOOR FINISHING
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=200 ..... FEED RATE FOR MILLING
13 L M99

14 CYCL DEF 24.0 SIDE FINISHING
  Q9=-1 ..... DIRECTION OF ROTATION
  Q10=-5 ..... PLUNGING DEPTH
  Q11=100 ..... FEED RATE FOR PLUNGING
  Q12=200 ..... FEED RATE FOR MILLING
  Q14=+0 ..... ALLOWANCE FOR SIDE
15 L M99

Retract tool, end
16 L Z+100 R0 F MAX M2
    
```

Solution:

FK Comb with 20-series cycles

SPGM

```
17 LBL 1
18 L X-80 Y-20 RL
19 FL Y+0 AN+90
20 FCT DR- R80 CCA+120 CCX+0 CCY+0
21 FCT DR- R8
22 FCT DR+ R64 CCX+0 CCY+0
23 FCT DR+ R8 CCY+8
24 FSELECT 2
25 FCT DR- R48 CCA+120 CCX+0 CCY+0
26 FCT DR- R8
27 FCT DR+ R32 CCX+0 CCY+0
28 FCT DR+ CCY+8
29 FCT X+0 Y+20 DR- R20 CCA+90 CCX+0 CCY+0
30 FSELECT 2
31 FLT X+12 AN+0
32 FL Y+0 AN-60
33 RND R4
34 FL AN+0
35 RND R4
36 FL AN+120 PAR31 DP16
37 FCT DR- R8 ICCA-180 CCY+34
38 FLT Y+0
39 RND R4
40 FL AN+0
41 RND R4
42 FL PAR39 DP16
43 FSELECT 2
44 FCT DR- R8 ICCA-180 CCY+64
45 FLT Y+0
46 RND R4
47 FL X+100 AN+0
48 FL Y-20 AN+270
49 FL X-80 AN+180
50 LBL 0

51 LBL 2
52 L X-90 Y-30 RR
53 L Y+90
54 L X+110
55 L Y-30
56 L X-90
57 LBL 0
58 END PGM 7176 MM
```



HEIDENHAIN

Advanced course TNC 4xx

C05



7176/4

Program layout: **FK Comb first as island and then mirrored in Y with SL I roughing**

Preparation

*BLK FORM
TOOL 1...*

Listing

*CYCL DEF 14 LABEL 1/2
CYCL DEF 6 ROUGH OUT*

Part contours

Machining cycle

*L X... Y... R0 F9999 M3
L Z... M99*

Execution

Upper half

*CYCL DEF 7 DATUM SHIFT
X0 Y-...*

*CYCL DEF 8 MIRROR IMAGE Y
CYCL CALL*

Y-values mirrored in lower half

Retract tool, end

L Z20 F9999 M2

**SPGM,
part contours**

LBL 1

Original island

*L X... Y... RL
:
:
:*

LBL 0

LBL 2

Original pocket


*L X... Y... RR
:
:
:*

LBL 0



Solution:

FK Comb first as island and then mirrored in Y



Main program

```
0 BEGIN PGM 75122 MM
1 BLK FORM 0.1 Z X-80 Y-95 Z-50
2 BLK FORM 0.2 X+100 Y+75 Z+0
3 TOOL DEF 1 L+0 R+3
4 TOOL CALL 1 Z S200
5 CYCL DEF 14.0 CONTOUR
6 CYCL DEF 14.1 CONTOUR LABEL 1/2
7 CYCL DEF 6.0 ROUGH OUT
8 CYCL DEF 6.1 SET UP -2 DEPTH -30
9 CYCL DEF 6.2 PECKG -10 F100 ALLOW +0
10 CYCL DEF 6.3 ANGLE +0 F200
11 L X-100 Y+0 R0 F MAX M3
12 L Z+2 F9999 M99
13 CYCL DEF 7.0 DATUM SHIFT
14 CYCL DEF 7.1 X+0
15 CYCL DEF 7.2 Y-20
16 CYCL DEF 8.0 MIRROR IMAGE
17 CYCL DEF 8.1 Y
18 CYCL CALL

19 CYCL DEF 8.0 MIRROR IMAGE
20 CYCL DEF 8.1
```

Retract tool, End

```
21 L Z+20 F MAX M2
```



HEIDENHAIN

Advanced course TNC 4xx

C01



75122/3

Solution:

FK Comb first as island and then mirrored in Y

SPGM

```
22 LBL 1
23 L X-80 Y+0 RL
24 FPOL X+0 Y+0
25 FC DR- R80 CCX+0 CCY+0
26 FCT DR- R8 CCPR+72 CCPA+120
27 FCT DR+ R64 CCX+0 CCY+0
28 FCT DR+ R8 CCY+8
29 FSELECT 2
30 FCT DR- R48 CCX+0 CCY+0
31 FCT DR- R8 CCPR+40 CCPA+120
32 FCT DR+ R32 CCX+0 CCY+0
33 FCT DR+ R6 CCY+6
34 FSELECT 2
35 FCT X+0 Y+20 DR- R20 CCX+0 CCY+0
36 FLT X+12 Y+20
37 FL AN-60
38 FCT Y+0 DR+ R4
39 FLT AN+0
40 FCT DR+ R4
41 FLT AN+120 PAR37 DP16
42 FCT DR- R8 CCY+34
43 FLT AN-60 PAR41 DP16
44 FCT Y+0 DR+ R4
45 FLT Y+0 AN+0
46 FCT DR+ R4
47 FLT AN+120 PAR43 DP16
48 FCT DR- R8 CCY+64
49 FLT AN-60 PAR47 DP16
50 FCT X+100 Y+0 DR+ R4
51 FSELECT 2
52 FL X+100 Y-20
53 FL X-80 Y-20
54 FL X-80 Y+0
55 LBL 0

56 LBL 2
57 L X-90 Y-20 RR
58 L Y+100
59 L X+120
60 L Y-20
61 L X-90
62 LBL 0
63 END PGM 75122 MM
```




HEIDENHAIN

Advanced course TNC 4xx

C05



75122/4



Program layout: **Multiple use of an island**

Preparation

```
BLK FORM / TOOL ...
CYCL DEF 14 CONTOUR
LABEL 10/2/3/4/5
CYCL DEF 20.0 CONTOUR DATA
CYCL DEF 22.0 ROUGH OUT
L X... Y... M3
L Z... M99
```

Without LBL 1 =
Original

Call

Retract tool, end

```
L Z100 M2
```

Execution

**SPGM,
part contours**

```
LBL 10 ... RR ... LBL 0
```

Pocket as outer limit

```
LBL 1
DATUM SHIFT = 0
ROTATION = 0 / SCALING = 1
MIRROR IMAGE = NO ENT
LBL 0
```

Island 1, original, not
used directly
Cycles reset

```
LBL 2
DATUM SHIFT = ...
CALL LBL 1
LBL 0
```

Island 2

```
LBL 3
DATUM SHIFT / SCALING = ...
CALL LBL 1
LBL 0
```

Island 3

```
LBL 4
DATUM SHIFT / ROTATION = ...
CALL LBL 1
LBL 0
```

Island 4


```
LBL 5
DAT. SHIFT / MIRROR IMAGE ...
CALL LBL 1
LBL 0
```

Island 5



Solution:

Multiple use of the same island-contour



Main program

```

0 BEGIN PGM 7177 MM
1 ..... MULTIPLE USE OF THE SAME
2 ..... ISLAND-CONTOUR
3 ..... IN A POCKET
4 BLK FORM 0.1 Z X+0 Y+0 Z-20
5 BLK FORM 0.2 X+100 Y+80 Z+0
6 TOOL DEF 1 L+0 R+5
7 TOOL CALL 1 Z S3000
8 CYCL DEF 14.0 CONTOUR GEOMETRY
9 CYCL DEF 14.1 CONTOUR LABEL 10 /2 /3 /4 /5
10 CYCL DEF 20.0 CONTOUR DATA
    Q1=-10 ..... MILLING DEPTH
    Q2=1 ..... TOOL PATH OVERLAP
    Q3=+0 ..... ALLOWANCE FOR SIDE
    Q4=+0 ..... ALLOWANCE FOR FLOOR
    Q5=+0 ..... WORKPIECE SURFACE COORD.
    Q6=-2 ..... SET UP CLEARANCE
    Q7=+20 ..... CLEARANCE HEIGHT
    Q8=+0 ..... ROUNDING RADIUS
    Q9=-1 ..... DIRECTION OF ROTATION
11 CYCL DEF 22.0 ROUGH OUT
    Q10=-5 ..... PLUNGING DEPTH
    Q11=100 ..... FEED RATE FOR PLUNGING
    Q12=500 ..... FEED RATE FOR MILLING
12 L X+10 Y+10 R0 F MAX M3
13 L Z+20 F9999 M99

Retract tool, end
14 L Z+100 F MAX M2
    
```



HEIDENHAIN

Advanced course TNC 4xx

C05



7177/3

Solution:

Multiple use of the same island-contour

SPGM 10, Pocket

```
15 LBL 10
16 L X-5 Y-5 RR
17 L X-5 Y+85
18 L X+105 Y+85
19 L X+105 Y-5
20 L X-5 Y-5
21 LBL 0
```

SPGM 1, Original

```
22 LBL 1
23 L X+0 Y+0 RL
24 L X+18 Y+20
25 L X+18 Y+0
26 L X+0 Y+0
27 CYCL DEF 7.0 DATUM SHIFT ..... RESET ALL
28 CYCL DEF 7.1 X+0
29 CYCL DEF 7.2 Y+0
30 CYCL DEF 10.0 ROTATION
31 CYCL DEF 10.1 ROT+0
32 CYCL DEF 11.0 SCALING
33 CYCL DEF 11.1 SCL1
34 CYCL DEF 8.0 MIRROR IMAGE
35 CYCL DEF 8.1
36 LBL 0

37 LBL 2
38 CYCL DEF 7.0 DATUM SHIFT
39 CYCL DEF 7.1 X+10
40 CYCL DEF 7.2 Y+10
41 CALL LBL 1
42 LBL 0

43 LBL 3
44 CYCL DEF 7.0 DATUM SHIFT
45 CYCL DEF 7.1 X+45
46 CYCL DEF 7.2 Y+5
47 CYCL DEF 11.0 SCALING
48 CYCL DEF 11.1 SCL1,5
49 CALL LBL 1
50 LBL 0

51 LBL 4
52 CYCL DEF 7.0 DATUM SHIFT
53 CYCL DEF 7.1 X+60
54 CYCL DEF 7.2 Y+45
55 CYCL DEF 10.0 ROTATION
56 CYCL DEF 10.1 ROT+15
57 CALL LBL 1
58 LBL 0

59 LBL 5
60 CYCL DEF 7.0 DATUM SHIFT
61 CYCL DEF 7.1 X+35
62 CYCL DEF 7.2 Y+55
63 CYCL DEF 8.0 MIRROR IMAGE
64 CYCL DEF 8.1 X
65 CALL LBL 1
66 LBL 0
67 END PGM 7177 MM
```



HEIDENHAIN

Advanced course TNC 4xx


C05



7177/4

Task: **Toggle joint**

Program(s): _____



Procedure:

- 2½ D-cut in Z/X
- Setting in Y

Tools:

- End mill R10 for roughing-out
- End mill R10 for finishing

Program layout:

Toggle joint with 1 tool, constant climb-cut

Preparation

Roughing-out

```
BLK FORM
TOOL DEF ... R10
TOOL 2 ...
L Z50 R0 F ... M3
CYCL DEF 7.0 DATUM SHIFT
CYCL DEF 7.1 X+Q108 Y0 / Z0
```

```
CALL LBL 1
```

Right

```
CYCL DEF 7.0 DATUM SHIFT
CYCL DEF 7.1 X-Q108 Y100 / Z0
CYCL DEF 10.0 ROT180
```

```
CALL LBL 1
```

Left

Retract tool, end

```
L Z ... M2
```

Contour Z/X, SPGM

```
LBL 1
```

```
L X+55 Y-4 R0 F ... M3
L Z ...
```

Starting plane
Starting Z

```
LBL 2
```

```
L IY ... F100
L X ... Z ... F250
L X ... Z ...
CT X ... Z ...
L Z ... F9999
L X ... Z ...
L Z ... F ...
```

Y-steps

Tangent
Arc

Retract tool
For further steps

```
CALL LBL 2 REP ...
```

```
L Z+ ...
CYCL DEF 7.0 DATUM SHIFT
CYCL DEF ... X0 Y0 Z0
CYCL DEF 10.0 ROT 0
```


Retract tool
Reset

```
LBL 0
```



Solution:

Toggle joint with 1 tool, constant climb-cut



Main program

```
0 BEGIN PGM 761260 MM
1 BLK FORM 0.1 Z X-50 Y+0 Z-52
2 BLK FORM 0.2 X+50 Y+100 Z+0
3 L Z+50 R0 F9999 M6
4 TOOL DEF 2 L+0 R+10
5 TOOL CALL 2 Z S4000
6 L Z+50 R0 F MAX M3
7 CYCL DEF 7.0 DATUM SHIFT
8 CYCL DEF 7.1 X+Q108
9 CYCL DEF 7.2 Y+0
10 CALL LBL 1 ..... RIGHT

11 CYCL DEF 7.0 DATUM SHIFT
12 CYCL DEF 7.1 X-Q108
13 CYCL DEF 7.2 Y+100
14 CYCL DEF 10.0 ROTATION
15 CYCL DEF 10.1 ROT+180
16 CALL LBL 1 ..... LEFT

Retract tool, end 17 L Z+50 R0 F MAX M2
```



HEIDENHAIN

Advanced course TNC 4xx

C01



761260/3

Solution: **Toggle joint with 1 tool,
constant climb-cut**

SPGM, Contour Z/X 18 LBL 1
19 L X+55 Y-4 R0 F9999 M3
20 L Z-51 F1000

21 LBL 2
22 L IY+2 F100
23 L X+39 Z-51 F250
24 L X+28,354 Z-20,2
25 CT X+0 Z+0
26 L Z+5 R0 F9999
27 L X+55 Z+5
28 L Z-51 F1000
29 CALL LBL 2 REP 50/50

30 L Z+5 R0 F MAX
31 CYCL DEF 7.0 DATUM SHIFT
32 CYCL DEF 7.1 X+0
33 CYCL DEF 7.2 Y+0
34 CYCL DEF 7.3 Z+0
35 CYCL DEF 10.0 ROTATION
36 CYCL DEF 10.1 ROT+0
37 LBL 0
38 END PGM 761260 MM



Program layout:

Toggle joint with 2 tools, constant climb-cut expanded roughing

Preparation
Roughing

```
BLK FORM  
TOOL DEF ... R10  
TOOL 1 ...  
FN 1: Q1 = Q108 + 5  
L Z50 R0 F ... M3  
CYCL DEF 7.0 DATUM SHIFT  
CYCL DEF ... X+Q1 Y0 Z+5
```

X > tool radius

```
CALL LBL 1
```

```
CYCL DEF 7.0 DATUM SHIFT  
CYCL DEF ... X-Q1 Y100 Z+5  
CYCL DEF 10.0 ROT180
```

```
CALL LBL 1
```

Completed section
already exists

```
TOOL 2 ...  
•  
•
```

```
CALL LBL 1
```

```
•  
•
```

```
CALL LBL 1
```

Retract tool, end

```
L Z ... M2
```

Contour Z/X,
SPGM 1
already exists

```
LBL 1
```

```
•  
•  
•  
•  
•  
•  
•  
•
```

```
LBL 0
```



Solution:

Toggle joint with 2 tools, constant climb-cut

Main program

Tool 1

```
0 BEGIN PGM 761261 MM
1 BLK FORM 0.1 Z X-50 Y+0 Z-52
2 BLK FORM 0.2 X+50 Y+100 Z+0
3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S2800
5 FN 1: Q1 = +Q108 + +5
6 L Z+50 R0 F9999 M3
7 CYCL DEF 7.0 DATUM SHIFT
8 CYCL DEF 7.1 X+Q1
9 CYCL DEF 7.2 Y+0
10 CYCL DEF 7.3 Z+5
11 CALL LBL 1 ..... RIGHT

12 CYCL DEF 7.0 DATUM SHIFT
13 CYCL DEF 7.1 X-Q1
14 CYCL DEF 7.2 Y+100
15 CYCL DEF 7.3 Z+5
16 CYCL DEF 10.0 ROTATION
17 CYCL DEF 10.1 ROT+180
18 CALL LBL 1 ..... LEFT
```

Tool 2

```
19 TOOL DEF 2 L+0 R+10
20 TOOL CALL 2 Z S4000
21 L Z+50 R0 F MAX M6
22 CYCL DEF 7.0 DATUM SHIFT
23 CYCL DEF 7.1 X+Q108
24 CYCL DEF 7.2 Y+0
25 CALL LBL 1 ..... RIGHT

26 CYCL DEF 7.0 DATUM SHIFT
27 CYCL DEF 7.1 X-Q108
28 CYCL DEF 7.2 Y+100
29 CYCL DEF 10.0 ROTATION
30 CYCL DEF 10.1 ROT+180
31 CALL LBL 1 ..... LEFT
```

Retract tool, end

```
32 L Z+50 R0 F MAX M2
```

SPGM, Contour Z/X

```
33 LBL 1
34 L X+55 Y-4 R0 F9999 M3
35 L Z-51 F1000


36 LBL 2
37 L IY+2 F100
38 L X+39 Z-51 F250
39 L X+28,354 Z-20,2
40 CT X+0 Z+0
41 L Z+5 R0 F9999
42 L X+55 Z+5
43 L X+55 Z-51 F MAX
44 CALL LBL 2 REP 50/50

45 L Z+5 R0 F MAX
46 CYCL DEF 7.0 DATUM SHIFT
47 CYCL DEF 7.1 X+0
48 CYCL DEF 7.2 Y+0
49 CYCL DEF 7.3 Z+0
50 CYCL DEF 10.0 ROTATION
51 CYCL DEF 10.1 ROT+0
52 LBL 0
53 END PGM 761261 MM
```



Task: **Truncated cone, standing, 2½ D, external**


Program(s): _____



- Procedure:**
- 2½ D cut
 - Vertical position in X/Y
 - Setting in Z

- Tools:**
- End mill

Truncated cone, standing, 2½ D, horizontal, with full circles



Calculation of step width per revolution:

$$\Delta PR = \frac{\text{End radius} - \text{Starting radius}}{\text{Number of steps}}$$

$$\Delta Z = \frac{\text{End height} - \text{Starting height}}{\text{Number of steps}}$$

Example: Number of steps = 10

$$\Delta PR = \frac{35 - 15}{10} = \frac{20}{10} = 2$$

$$\Delta Z = \frac{-20 - 0}{10} = \frac{-20}{10} = -2$$



Program layout:

Truncated cone, standing, external, 2½ D, horizontal

Preparation

Roughing

```
BLK FORM  
TOOL 1... R10  
L Z20 R0 F9999 M3  
CC X... Y...  
LP PR65 PA0 R... F... M...  
L Z+2
```

Z-Clearance

```
LBL 1
```

```
LP IPR-... RL F...  
L Z-30  
CP IPA-360 DR- RL F... M98
```

```
CALL LBL 1 REP...
```

```
L Z... M6
```

Finishing

```
TOOL 2... R10  
L Z+... R0 F... M3  
LP PR15 PA0 R0 F...  
LP IPR+Q108
```

Compensate tool
radius outside
Contact

```
L Z0 F150
```

```
LBL 2
```

```
LP IPR+1  
L IZ-1 F200  
CP IPA-360 DR- F1000
```

} Cone angle 45°
for slope
Full circle in X/Y

```
CALL LBL 2 REP...
```


Retract tool, end

```
L Z20... M2
```



Solution:

Truncated cone, standing, 2½ D, horizontal, external form with end mill



```
0 BEGIN PGM 7162 MM
1 ..... TRUNCATED CONE STANDING
2 ..... 2½ D, HORIZONTAL, END MILL
3 BLK FORM 0.1 Z X+0 Y+10 Z-31
4 BLK FORM 0.2 X+100 Y+100 Z+0
5 TOOL DEF 1 L+0 R+10
6 TOOL CALL 1 Z S2000
7 CC X+50 Y+50
8 L Z+20 R0 F9999 M3
9 LP PR+65 PA+0
10 L Z+2

11 LBL 1
12 LP IPR-5 RL F200
13 L Z-30
14 CP IPA-360 DR- RL F300 M98
15 CALL LBL 1 REP 5/5
16 L Z+20 R0 F9999 M6


17 TOOL DEF 2 L+0 R+10
18 TOOL CALL 2 Z S2800
19 L Z+20 R0 F9999 M3
20 LP PR+15 PA+0 R0 F9999 M3
21 LP IPR+Q108
22 L Z+2
23 L Z+0 F150

24 LBL 2
25 LP IPR+1
26 L IZ-1 F200
27 CP IPA-360 DR- F1000 M98
28 CALL LBL 2 REP 19/19

29 L Z+20 R0 F9999 M2
30 END PGM 7162 MM
```

Task: **Truncated cone, standing, 3D,
external form**

Program(s): _____




- Procedure:**
- 3D-cut from top to bottom
 - Rotation about zero point at center of truncated cone

- Tools:**
- End mill R6

Detail:

Truncated cone, standing, 3D, external form with end mill, radius compensation



Program layout:

Truncated cone, standing, 3D, external form

Preparation

```
BLK FORM  
TOOL ...  
L Z ... M3  
CYCL DEF 7.0 DATUM SHIFT ...  
Q3 = 15 + Q108  
Q4 = 35 + Q108
```

Dat. sh. in center of part
Tool compensation

1 call

```
CALL LBL 1
```

Retract tool, end

```
L Z ... M2
```

1 revolution, SPM

```
LBL 1
```

```
L XQ3 Y0 R0 F9999 M3
```

Approach compensated
X-value in drawing

```
L Z2  
L Z0 F100
```

Contact

```
LBL 2
```

```
L XQ4 Y0 Z-20
```

3D-cut, mill stays free!
Feed rate for first cut
=100
Z return

```
L Z0 F9999  
L XQ3 Y0 F2000  
CYCL DEF 10.0 ROTATION  
CYCL DEF 10.1 IROT-5  
L Y0 F500
```

Feed rate for
following cuts

```
CALL LBL 2 REP ...
```


```
CYCL DEF 10.0 ROTATION  
CYCL DEF 10.1 ROT+0
```

```
LBL 0
```



Solution:

Truncated cone, standing, 3D, vertical, external form with end mill, repeatable



Main program

```
0 BEGIN PGM 7282 MM
1 ..... TRUNCATED CONE, STANDING,
2 ..... 3D, VERTICAL, END MILL
3 ..... COMPLETED SECTION
4 BLK FORM 0.1 Z X+10 Y+10 Z-32
5 BLK FORM 0.2 X+90 Y+90 Z+0
6 TOOL DEF 1 L+0 R+10
7 TOOL CALL 1 Z S4000
8 L Z+20 R0 F9999 M3
9 CYCL DEF 7.0 DATUM SHIFT
10 CYCL DEF 7.1 X+50
11 CYCL DEF 7.2 Y+50
12 FN 1: Q3 = +15 + +Q108 ..... X TOP
13 FN 1: Q4 = +35 + +Q108 ..... X BOTTOM
14 CALL LBL 1 ..... FIRST CUT

15 L Z+50 R0 F9999 M2
```

SPGM

```
16 LBL 1
17 L X+Q3 Y+0 R0 F9999 M3 ..... APPROACH TOP
18 L Z+2
19 L Z+0 F100

20 LBL 2
21 L X+Q4 Y+0 Z-20 ..... CUT DOWNWARDS
22 L Z+0 F9999 ..... Z FREE
23 L X+Q3 Y+0 F2000
24 CYCL DEF 10.0 ROTATION
25 CYCL DEF 10.1 IROT-5 ..... FINENESS OF ROTATION
26 L Y+0 F500 ..... CROSS OVER ON TOP
27 L F500
28 CALL LBL 2 REP 71/71 ..... NUMBER OF ROTATION STEPS

29 CYCL DEF 10.0 ROTATION
30 CYCL DEF 10.1 ROT+0
31 LBL 0
32 END PGM 7282 MM
```




HEIDENHAIN

Advanced course TNC 4xx

C05



7282/4




- Procedure:**
- 3 D-cut from bottom to top
 - Pre-machining with 2 circular pockets

- Tools:**
- End mill R6

Solution:

Truncated cone, standing, 3D, vertical, internal form with end mill, upwards



Main program

```
0 BEGIN PGM 7283 MM
1 ..... TRUNCATED CONE, STANDING,
2 ..... 3D, VERTICAL, INTERNAL, END
3 ..... MILL, COMPLETED SECTION, B06
4 BLK FORM 0.1 Z X+0 Y+10 Z-32
5 BLK FORM 0.2 X+100 Y+90 Z+0
6 TOOL DEF 1 L+0 R+10
7 TOOL CALL 1 Z S2800
8 L Z+20 R0 F9999 M3
9 CYCL DEF 7.0 DATUM SHIFT
10 CYCL DEF 7.1 X+50
11 CYCL DEF 7.2 Y+50
12 FN 1: Q3 = +35 + -Q108
13 FN 1: Q4 = +15 + -Q108
14 CALL LBL 1

15 L Z+50 R0 F9999 M2
```

SPGM

```
16 LBL 1 ..... VERY FIRST CUT
17 L X+10 Y+0
18 L Z-20 R0 F400 M3
19 L X+Q3 Y+0

20 LBL 2 ..... FOLLOWING CUTS 3D UPWARDS
21 L X+Q3 Y+0 Z-20 F1000
22 L X+Q4 F9999
23 L Z-30 F2000
24 CYCL DEF 10.0 ROTATION
25 CYCL DEF 10.1 IROT-5
26 CALL LBL 2 REP 71/71

27 CYCL DEF 10.0 ROTATION
28 CYCL DEF 10.1 ROT+0
29 LBL 0
30 END PGM 7283 MM
```



HEIDENHAIN

Advanced course TNC 4xx

C01



7283/3

Solution:

Truncated cone, standing, 3D, vertical, internal form with end mill, upwards, with roughing

Main program

```
0 BEGIN PGM 7284 MM
1 ..... TRUNCATED CONE, STANDING,
2 ..... VERTICAL, INTERNAL FORM
3 ..... FROM SOLID BLOCK WITH
4 ..... END MILL
5 BLK FORM 0.1 Z X+0 Y+10 Z-32
6 BLK FORM 0.2 X+100 Y+90 Z+0
7 TOOL DEF 1 L+0 R+10
8 TOOL CALL 1 Z S2800
9 L Z+20 R0 F9999 M3
10 FN 0: Q1 = +20 ..... DEPTH
11 FN 0: Q2 = +35 ..... RADIUS
12 CYCL DEF 5.0 CIRCULAR POCKET
13 CYCL DEF 5.1 SET UP-2
14 CYCL DEF 5.2 DEPTH -Q1
15 CYCL DEF 5.3 PECKG -5 F150
16 CYCL DEF 5.4 RADIUS Q2
17 CYCL DEF 5.5 F500 DR+
18 L X+50 Y+50
19 L Z+2 M99 ..... ROUGHING OUT
20 FN 0: Q1 = +10
21 FN 0: Q2 = +15

22 L Z-18 M99
23 CYCL DEF 7.0 DATUM SHIFT
24 CYCL DEF 7.1 X+50
25 CYCL DEF 7.2 Y+50
26 FN 1: Q3 = +15 + -Q108 ..... XA BOTTOM
27 FN 1: Q4 = +35 + -Q108 ..... XB TOP
28 FN 0: Q5 = -25 ..... Z DEPTH, FIRST CUT

29 CALL LBL 1 ..... ROUGHING OUT SLOPE

30 FN 0: Q5 = -30 ..... Z DEPTH, SECOND CUT

31 CALL LBL 1 ..... FINISHING

32 L Z+50 R0 F9999 M2
```

SPGM, Contour


```
33 LBL 1
34 L X+Q3 Y+0 Z+Q5 R0 F400 M3 ..... A
35 L X+Q4 Y+0 Z+0 ..... B

36 LBL 2
37 L X+Q3 Y+0
38 L Z+Q5
39 L X+Q4 Y+0 Z+0 F1000 ..... B
40 L X+Q3 F9999 ..... A
41 L Z+Q5 F2000
42 CYCL DEF 10.0 ROTATION
43 CYCL DEF 10.1 IROT-5
44 CALL LBL 2 REP 71/71
45 CYCL DEF 10.0 ROTATION
46 CYCL DEF 10.1 ROT+0
47 LBL 0
48 END PGM 7284 MM
```



Task: **Hemisphere, internal machining,
2D, horizontal**

Program(s): _____



Procedure:

- Full circles (vertical positions) in X/Y
- Pre-machine with 2 circular pockets

Tools:

- Roughing out (circular pockets) with end mill R10
- Machine contour with spherical mill R5



HEIDENHAIN

Advanced course TNC 4xx


C05



7322/1
7323/1

Program layout:

Hemisphere, internal machining with spherical mill conventional with R0, 2D, horizontal



Program layout:

**Hemisphere, internal machining with spherical mill
conventional with R0, 2D, horizontal, working
downwards**

Preparation

```
BLK FORM  
TOOL 1 ... R5  
L Z10 R0 F9999 M3
```

Approach

```
CC X50 Y50  
LP PR50 PA0 R0 F9999  
LP IPR-Q108  
L Z+0 R0 F100
```

Z-clearance

X/Y plane

$PR = R_{\text{Part}} - R_{\text{Tool}}$

Account for tool radius

Plunge

```
LBL 1
```

Full circles

```
CC X50 Y50  
CP DR+
```

X/Y plane

Setting

```
CC X50 Z0
```

Z/X plane

Angular steps

```
CP IPA2 DR+
```


```
CALL LBL 1 REP 44/44
```

```
L Z+ ... M2
```



Solution:

Hemisphere, internal machining with spherical mill conventional with R0, 2D, horizontal




```

0 BEGIN PGM 7322 MM
1 BLK FORM 0.1 Z X-2 Y-2 Z-55
2 BLK FORM 0.2 X+102 Y+102 Z+0
3 TOOL DEF 1 L+0 R+5
4 TOOL CALL 1 Z S4000
5 L Z+70 R0 F9999 M6 ..... Z-CLEARANCE
6 CC X+50 Y+50 ..... CENTER IN X/Y
7 LP PR+50 PA+0 R0 F9999 M3 ..... STARTING POINT
8 LP IPR-Q108 ..... ACCOUNT FOR TOOL RADIUS
9 L Z+0 R0 F100

10 LBL 1
11 CC X+50 Y+50 ..... CENTER IN X/Y
12 CP DR+ F500 ..... CIRCULAR RING
13 CC X+50 Z+0 ..... CENTER IN Z/X
14 CP IPA+2 DR+ F100 ..... MOVE DEEPER (ARC)
15 CALL LBL 1 REP 44/44 ..... FURTHER REVOLUTIONS

16 L Z+70 R0 F9999 M2
17 END PGM 7322 MM
  
```

Set datum



HEIDENHAIN

Advanced course TNC 4xx

C05



7322/4

Solution:

Hemisphere, internal machining with end and die sink mill

Main program

```
0 BEGIN PGM 7323 MM
1 BLK FORM 0.1 Z X-2 Y-2 Z-51
2 BLK FORM 0.2 X+102 Y+102 Z+0
```

End mill

```
3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S4000
5 FN 0: Q1 = +45 ..... UPPER POCKET RADIUS
6 CALL LBL 1 ..... ROUGHING-OUT
7 L Z+100 R0 F MAX M6 ..... TOOL CHANGE
```

Die sink mill

```
8 TOOL DEF 2 L+0 R+5
9 TOOL CALL 2 Z S4500
10 CALL LBL 2 ..... FINISHING
```

Retract tool, end

```
11 L Z+100 R0 F MAX M2
```

SPGM 1, Roughing out with end mill

```
12 LBL 1
13 CC X+50 Y+50 ..... CENTER IN X/Y
14 LP PR+0 PA+0 R0 F MAX M3
15 CYCL DEF 5.0 CIRCULAR POCKET
16 CYCL DEF 5.1 SET UP -2
17 CYCL DEF 5.2 DEPTH -14
18 CYCL DEF 5.3 PECKG -5 F150
19 CYCL DEF 5.4 RADIUS Q1
20 CYCL DEF 5.5 F500 DR+
21 L Z+2 R0 F MAX M99
22 FN 0: Q1 = +30 ..... LOWER POCKET RADIUS
23 L Z-12 R0 F MAX M99
24 LBL 0
```

SPGM 2, Finishing with die sink mill (spherical form)

```
25 LBL 2
26 LP PR+50 PA+0 R0 F500 M3 ..... STARTING POINT
27 LP IPR-Q108
28 L Z+0 R0 F MAX


29 LBL 20
30 CC X+50 Y+50 ..... CENTER IN X/Y
31 C DR+ F500 ..... CIRCULAR RING
32 CC X+50 Z+0 ..... CENTER IN Z/X
33 CP IPA+2 DR+ F150 ..... MOVE DEEPER
34 CALL LBL 20 REP 44/44 ..... FURTHER CIRCULAR RINGS

35 LBL 0
36 END PGM 7323 MM
```



Task: **Hemisphere, external machining
with spherical mill, working
downwards, $2\frac{1}{2} D$, horizontal**

Program(s): _____



Procedure:

- Full circles (vertical positions) in X/Y
- Setting with angular steps in Z/X
- Pre-machining e.g. with contour island or helical interpolation

Tools:

- (Roughing out with end mill R5)
- Finishing with spherical mill R5



Program layout:

**Hemisphere, external machining with spherical mill,
working downwards, 2½ D, horizontal**

Preparation

Finishing

```
BLK FORM  
TOOL 1...R5  
L Z 20 R0 F9999 M3
```

Call

```
CALL LBL 1
```

Retract tool, end

```
L Z...M2
```

SPGM

```
LBL 1
```

```
L X+50 Y+50 R...F...M...  
L Z+Q108 F100
```

Approach center
Account for tool
radius

```
LBL 2
```

```
CC Z-35 X+50  
CP IPA+2 DR+ F100
```

} Setting in Z/X

```
CC X+50 Y+50  
CP DR- F500
```

} Full circle in X/Y

```
CALL LBL 2 REP...
```


Further cuts

```
LBL 0
```



Solution:

Hemisphere, external machining with spherical mill, working downwards, $2\frac{1}{2} D$, horizontal



Main program

```
0 BEGIN PGM 72841 MM
1 ..... EXTERNAL HEMISPHERE 2D,
2 ..... SPHERICAL MILL,
3 ..... MILLING DOWNWARDS
4 BLK FORM 0.1 Z X+0 Y+10 Z-36
5 BLK FORM 0.2 X+100 Y+90 Z+5
6 TOOL DEF 1 L+0 R+5
7 TOOL CALL 1 Z S2800
8 L Z+20 R0 F9999 M3
9 CALL LBL 1
```

Retract tool, end

```
10 L Z+100 R0 F9999 M2
```

SPGM

```
11 LBL 1
12 L X+50 Y+50 ..... CENTER IN X/Y
13 L Z+Q108 F100 ..... CONTACT WITH TOOL RADIUS

14 LBL 2
15 CC Z-35 X+50
16 CP IPA+2 DR+ F100 ..... SETTING
17 CC X+50 Y+50
18 CP DR- F500 ..... FULL CIRCLE
19 CALL LBL 2 REP 44/44
20 LBL 0
21 END PGM 72841 MM
```



HEIDENHAIN

Advanced course TNC 4xx


C05



72841/3

Task: **Hemisphere, external machining
with spherical mill, only from
below, 3D**

Program(s): _____



Procedure:

- 3D-cut vertical
- Always from below

Tools:

- End mill R10



HEIDENHAIN

Advanced course TNC 4xx


C09



76130/1
76131/1

Program layout:

Hemisphere, external machining with spherical mill, only from below, 3D, 1 cut



HEIDENHAIN

Advanced course TNC 4xx

C01



76130/2

Program layout:

**Hemisphere, external machining with spherical mill,
only from below, 3D, 1 cut**

Preparation

Pre-position

“Virtual
circular arc”

<i>BLK FORM</i>	
<i>TOOL 1 . . .</i>	
<i>L Z50 R0 F MAX M6</i>	
<i>L X120 Y50 R0 F MAX M3</i>	
<i>CYCL DEF 7.0 DATUM SHIFT</i>	
<i>X50 Y+50 Z-50</i>	
<i>L Y0 Z0 R0 F MAX</i>	
<i>LBL 2</i>	
<i>CC Z . . . X . . .</i>	
<i>L Y0 . . .</i>	
<i>LP PR50 PA90 R0 . . .</i>	
<i>LBL 3</i>	
<i>LP IPA-5</i>	
<i>CALL LBL 3 REP 17/17</i>	
<i>L X50 . . .</i>	
<i>L X65 Z0 . . .</i>	
<i>CYCL DEF 10.0 ROTATION</i>	
<i>CYCL DEF 10.1 IROT+5</i>	
<i>CALL LBL 2 REP 53/53</i>	
<i>CYCL DEF 7.0 DATUM SHIFT = 0</i>	
<i>CYCL DEF 10.0 ROTATION = 0</i>	
<i>L Z10 R0 F9999</i>	

Zero point in center
of sphere

Compensate circle ctr.

Contour values

1 angular step
(angle in space)

Retract tool

1 further arc


Reset

Retract tool



Solution:

Hemisphere, external machining with spherical mill, only from below, 3D, 1 cut



```

0 BEGIN PGM 76130 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-55
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S4000
5 L Z+50 R0 F MAX M6
6 L X+120 Y+50 R0 F MAX M3
7 CYCL DEF 7.0 DATUM SHIFT
8 CYCL DEF 7.1 X+50
9 CYCL DEF 7.2 Y+50
10 CYCL DEF 7.3 Z-50
11 L Y+0 Z+0 R0 F MAX ..... END OF DEPTH
                                (CENTER OF SPHERE)

12 LBL 2
13 CC Z+0 X+Q108 ..... SYSTEMATIC SHIFT IN X
14 L Y+0 F200
15 LP PR+50 PA+90 R0 F200 ..... SPHERE RADIUS

16 LBL 3
17 LP IPA-5 ..... 1 ANGULAR STEP
                                (ANGLE IN SPACE)

18 CALL LBL 3 REP 17/17

19 L X+50 Y+0 R0 F2000
20 L X+65 Z+0 F1000
21 CYCL DEF 10.0 ROTATION ..... 1 FURTHER ARC
22 CYCL DEF 10.1 IROT+5
23 CALL LBL 2 REP 53/53

24 CYCL DEF 7.0 DATUM SHIFT
25 CYCL DEF 7.1 X+0
26 CYCL DEF 7.2 Y+0
27 CYCL DEF 7.3 Z+0
28 CYCL DEF 10.0 ROTATION
29 CYCL DEF 10.1 ROT+0
30 L Z+20 R0 F MAX
31 END PGM 76130 MM

```

Program layout:

**Hemisphere, external machining with spherical mill,
only from below, 3D, 2 cuts**

Preparation

```
BLK FORM.../TOOL 1...R 10  
L X... Y... R0 F9999 M3  
CYCL DEF 7.0 DATUM SHIFT  
X50 Y50 Z-50  
L Y0 Z15...
```

External pre-positioning

Center of sphere

Roughing

```
LBL 1
```

```
CC Z15 XQ108  
L Y0 F200  
LP PR50 PA90...
```

Z oversize

Tilt plane

Contour values

```
LBL 2
```

```
LP IPA-10 F500
```

1 arc, coarse, not to
largest angle

```
CALL LBL 2 REP 3/3
```

```
L X50 Y0 F MAX  
L X65 Z15 F1000  
CYCL DEF 10 ROTATION IROT+18
```

Z oversize

```
CALL LBL 1 REP...
```

Additional arcs

```
CYCL DEF 10 ROTATION ROT+0  
L Y0 Z0 F...
```

Finishing

```
LBL 3
```

```
CC Z0 XQ108  
L Y0  
LP PR50 PA90 F200
```

Z specified size

```
LBL 4
```

```
LP IPA-1 F800
```

1 arc, fine, to largest
angle in space

```
CALL LBL 4 REP 89/89
```

Return

```
L X+50 Y+0...  
L X+65 Z+0 F1000  
CYCL DEF 10 ROTATION IROT+1
```

```
CALL LBL 3 REP...
```

```
CYCL DEF 7 DATUM SHIFT = 0  
CYCL DEF 10 ROTATION = 0
```

Retract tool, end

```
L Z20... M2
```



HEIDENHAIN

Advanced course TNC 4xx


C05



76131/2

Solution:

Hemisphere, external machining with spherical mill, only from below, 3D, 2 cuts



Roughing

```
0 BEGIN PGM 76131 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-60
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+10
4 TOOL CALL 1 Z S4000
5 L Z+50 R0 F MAX M6
6 L X+120 Y+50 R0 F MAX M3
7 CYCL DEF 7.0 DATUM SHIFT
8 CYCL DEF 7.1 X+50
9 CYCL DEF 7.2 Y+50
10 CYCL DEF 7.3 Z-50
11 L Y+0 Z+15 R0 F MAX ..... Z OVERSIZE

12 LBL 1
13 CC Z+15 X+Q108 ..... Z OVERSIZE
14 L Y+0 F200
15 LP PR+50 PA+90 R0 F200 ..... SPHERE RADIUS

16 LBL 2 ..... 1 ARC COARSE
17 LP IPA-10 R0 F500
18 CALL LBL 2 REP 3/3

19 L X+50 Y+0 R0 F MAX
20 L X+65 Z+15 R0 F1000
21 CYCL DEF 10.0 ROTATION ..... 19 FURTHER ARCS
22 CYCL DEF 10.1 IROT+18
23 CALL LBL 1 REP 19/19

24 CYCL DEF 10.0 ROTATION
25 CYCL DEF 10.1 ROT+0
```



HEIDENHAIN

Advanced course TNC 4xx

C01



76131/3

Solution:

Hemisphere, external machining with spherical mill, only from below, 3D, 2 cuts

Finishing

```
26 L Y+0 Z+0 R0 ..... Z SPECIFIED SIZE
27 LBL 3
28 CC Z+0 X+Q108 ..... Z SPECIFIED SIZE
29 L Y+0
30 LP PR+50 PA+90 F200 ..... SPHERE RADIUS

31 LBL 4 ..... 1 ARC FINE
32 LP IPA-1 F800
33 CALL LBL 4 REP 89/89


34 L X+50 Y+0 F2000
35 L X+65 Z+0 F1000
36 CYCL DEF 10.0 ROTATION ..... 71 FURTHER ARCS
37 CYCL DEF 10.1 IROT+1
38 CALL LBL 3 REP 359/359

39 CYCL DEF 7.0 DATUM SHIFT
40 CYCL DEF 7.1 X+0
41 CYCL DEF 7.2 Y+0
42 CYCL DEF 7.3 Z+0
43 CYCL DEF 10.0 ROTATION
44 CYCL DEF 10.1 ROT+0
45 L Z+20 R0 F MAX
46 END PGM 76131 MM
```



Task: **Thread milling, in three steps
M64 x 1,5 inside, 2 settings**

Program(s): _____



Precondition:

- Center must be free!

Procedure:

- Helical interpolation in X/Y plane with simultaneous Z motion
- Working direction downwards

Tools:

- Threading mill with cutting edge corresponding to thread profile



Program layout:

**Thread milling, in three steps
M64 x 1,5 inside, 2 settings**

Preparation

*BLK FORM
TOOL 1 . . .
L Z+20 R0 F9999 M3
L X+50 Y+50 . . .
CC*

Retract tool
Approach center
Auto. take over of pole

**First cut
with oversize**

*LBL 1
L Z+0 . . . F500
LP PR+31 PA0 RR F150*

Contact in Z
Radius oversize

*LBL 2
CP IPA-360 IZ-4,5 DR- F500
CALL LBL 2 REP 4/4*

Helix

*LP PR0 PA0 R0 F9999
L Z+0*

**Second cut
to final size**

LP PR+32 PA0 RR F150

Final size, re-contact

*LBL 3
CP IPA-360 IZ-4,5 DR- F500
CALL LBL 3 REP 4/4*

Helix

*LP PR0 PA0 R0 F9999
CYCL DEF 10.0 ROTATION
CYCL DEF 10.1 IROT+120*

Center

CALL LBL 1 REP 2/2

Further threads


Retract tool, end

L Z20 . . . M2



Solution:

Thread milling, in three steps M64 x 1,5 inside, 2 settings



```
0 BEGIN PGM 7164 MM
1 BLK FORM 0.1 Z X+0 Y+10 Z-20
2 BLK FORM 0.2 X+100 Y+90 Z+0
3 TOOL CALL 1 Z S2000
4 TOOL DEF 1 L+0 R+20
5 L Z+20 R0 F9999 M3
6 L X+50 Y+50 R0 F9999 M3
7 CC
```

First step, beginning at 0°

```
8 LBL 1
9 L Z+0 R0 F9999
```

First cut, diameter 62

```
10 LP PR+31 PA+0 RR F500 ..... APPROACH WALL
```

```
11 LBL 2
12 CP IPA-360 IZ-4,5 DR- F500 ..... HELIX
13 CALL LBL 2 REP 4/4
```

```
14 LP PR+0 PA+0 R0 F9999 ..... APPROACH CENTER
15 L Z+0
```

Final cut, diameter 64

```
16 LP PR+32 PA+0 RR F500 ..... APPROACH WALL
```

```
17 LBL 3
18 CP IPA-360 IZ-4,5 DR- ..... HELIX
19 CALL LBL 3 REP 4/4
```

```
20 LP PR+0 PA+0 R0 F9999 ..... APPROACH CENTER
21 CYCL DEF 10.0 ROTATION ..... ROTATION
```

Second and third steps, each offset by 120°


```
22 CYCL DEF 10.1 IROT+120
23 CALL LBL 1 REP 2/2
```

```
24 L Z+20 R0 M2
25 END PGM 7164 MM
```



Task: **Milling a contour with several settings**

Program(s): _____



Q	Meaning	Value	Comment
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

Conventional preparation
Load data

```
BLK FORM

FN 0: Q1 = ...
FN 0: Q2 = ...
FN 0: Q3 = ...
FN 0: Q4 = ...
FN 0: Q5 = ...
FN 0: Q6 = ...

TOOL 1 ... R5
L Z+ ... R0 F9999 M3

CALL LBL 1
```

Feed rate Z
Feed rate plane
Start Z
End Z
Number of cuts
Rounding

Execute

Retract tool, end

```
L Z100 ... M2
```

SPGM

Starting values

Starting position

End position

```
LBL 1

FN 2: Q12 = Q4 - Q3
FN 4: Q12 = Q12 DIV Q5
FN 0: Q13 = Q3
L X-10 Y+70 R0 F9999 M3

LBL 2

L ZQ13 FQ1
L X+10 Y+70 RL FQ2
RND ...
L ... / RND ... / L ...
L X ... Y ... RL
RND ...
L X- ... Y ... R0 ...

FN 1: Q13 = Q13 + Q12

FN 11: IF Q13 GT Q4
        GOTO LBL 2

FN 9: IF Q13 EQU Q4
        GOTO LBL 2

L Z2 F9999

LBL 0
```

Setting range Z
Setting step Z
Z-start = Z-current
Uncorrected!

Move deeper
Move to first contour pt.
Approach tangentially


First/last contour point
Depart tangentially
Depart , uncorrected

Calculate Z setting

} Return jump /
further cuts, if the
final depth has not
yet been milled

Solution:

Milling a contour with several settings



Main program

```
0 BEGIN PGM 7285 MM
1 ..... MILLING A CONTOUR WITH
SEVERAL SETTINGS

2 BLK FORM 0.1 Z X-10 Y+0 Z-40
3 BLK FORM 0.2 X+100 Y+100 Z+0
4 FN 0: Q1 = +100 ..... FEED RATE Z
5 FN 0: Q2 = +500 ..... FEED RATE PLANE
6 FN 0: Q3 = -2 ..... START Z
7 FN 0: Q4 = -20 ..... END Z
8 FN 0: Q5 = +4 ..... NUMBER OF CUTS
9 FN 0: Q6 = +10 ..... ROUNDING
10 TOOL DEF 1 L+0 R+5
11 TOOL CALL 1 Z S2800
12 L Z+10 R0 F9998 M3
13 CALL LBL 1

Retract tool, end
14 L Z+100 R0 F9998 M2
```



Solution:

Milling a contour with several settings

SPGM


```
15 LBL 1
16 FN 2: Q12 = +Q4 - +Q3
17 FN 4: Q12 = +Q12 DIV +Q5 ..... Z SETTING
18 FN 0: Q13 = +Q3 ..... CURRENT Z
19 L X-10 Y+70 R0 F9998 M3

20 LBL 2
21 L Z+Q13 FQ1
22 L X+10 Y+70 RL FQ2
23 RND R2
24 L Y+90
25 RND RQ6
26 L X+50
27 RND RQ6
28 L X+90 Y+50
29 RND RQ6
30 L Y+10
31 RND RQ6
32 L X+50
33 RND RQ6
34 L X+10 Y+50
35 RND RQ6
36 L Y+70
37 RND R2
38 L X-10 Y+70 R0 F9998
39 FN 1: Q13 = +Q13 + +Q12
40 FN 11: IF +Q13 GT +Q4 GOTO LBL 2
41 FN 9: IF +Q13 EQU +Q4 GOTO LBL 2
42 L Z+2 F9998
43 LBL 0
44 END PGM 7285 MM
```



Task: **Finish milling rectangular pocket**

Program(s): _____



Q	Meaning	Value	Comment
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

Load data

```

:
:
FN 0: Q1 = ... Q9 = ...
BLK FORM
TOOL 1 ... R5
L Z10 R0 F9999 M3
CALL LBL 1
    
```

Retract tool, end

```
L Z ... M2
```

SPGM


```

LBL 1
FN 4: Q13 = Q3 DIV 2
FN 4: Q14 = Q4 DIV 2
FN 4: Q16 = Q108 DIV 4
FN 4: Q17 = Q7 DIV 2
L X+Q1 Y+Q2 R0 F9999 M3
L Z+Q9
L Z-Q5 FQ7
L IXQ13 YQ2 RL
RND RQ ... FQ ...
L IY ...
:
:
L IX0 IYQ14
RND RQ ... FQ ...
L XQ1 YQ2 R0 F9999
L Z+Q9
LBL 0
    
```

- X/2
- Y/2
- Approach radius (*RND*)
- Appr. feed rate (*RND*)
- Approach center
- Z-clearance
- Z-depth
- First contour point
- Last contour point
- Move to center
- Retract tool in Z

Solution:

Finish milling rectangular pocket



Main program

```
0 BEGIN PGM 7286 MM
1 FN 0: Q1 = +50 ..... X-CENTER
2 FN 0: Q2 = +50 ..... Y-CENTER
3 FN 0: Q3 = +90 ..... LENGTH X
4 FN 0: Q4 = +70 ..... WIDTH Y
5 FN 0: Q5 = +20 ..... DEPTH Z
6 FN 0: Q6 = +10 ..... ROUNDING
7 FN 0: Q7 = +1000 ..... FEED RATE Z
8 FN 0: Q8 = +300 ..... FEED RATE PLANE
9 FN 0: Q9 = +2 ..... Z-CLEARANCE
10 BLK FORM 0.1 Z X+0 Y+10 Z-40
11 BLK FORM 0.2 X+100 Y+90 Z+0
12 TOOL DEF 1 L+0 R+5
13 TOOL CALL 1 Z S2800
14 L Z+10 R0 F9998 M3
15 CALL LBL 1
```

Retract tool, end

```
16 L Z+50 R0 M2
```



HEIDENHAIN

Advanced course TNC 4xx

C01



7286/3


Solution:

Finish milling rectangle pocket

SPGM

```
17 LBL 1
18 FN 4: Q13 = +Q3 DIV +2
19 FN 4: Q14 = +Q4 DIV +2
20 FN 4: Q16 = +Q6 DIV +4
21 FN 4: Q17 = +Q7 DIV +2
22 L X+Q1 Y+Q2 R0 F9998 M3
23 L Z+Q9
24 L Z-Q5 FQ7
25 L IX+Q13 Y+Q2 RL
26 RND RQ16 FQ17
27 L IY+Q14 IX+0 FQ8
28 RND RQ6 FQ17
29 L IX-Q3 IY+0
30 RND RQ6 FQ17
31 L IY-Q4 IX+0
32 RND RQ6 FQ17
33 L IX+Q3 IY+0
34 RND RQ6 FQ17
35 L IY+Q14 IX+0
36 RND RQ16 FQ17
37 L X+Q1 Y+Q2 R0 F9998
38 L Z+Q9
39 LBL 0
40 END PGM 7286 MM
```






Q	Meaning	Value	Comment
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

Solution:

Milling cylindrical pins



Main program

```
0 BEGIN PGM 7288 MM
1 ..... MILLING CYLINDRICAL PINS
2 ..... WITH 2 CUTS AND
   ..... RADIAL SETTING

3 FN 0: Q1 = +50 ..... X-CENTER
4 FN 0: Q2 = +50 ..... Y-CENTER
5 FN 0: Q3 = +0 ..... Z-UPPER EDGE
6 FN 0: Q4 = +80 ..... DIAMETER
7 FN 0: Q5 = +1 ..... OVERSIZE RADIUS
8 FN 0: Q6 = +20 ..... MILL DEPTH
9 FN 0: Q7 = +200 ..... MILL FEED RATE
10 BLK FORM 0.1 Z X+0 Y+10 Z-22
11 BLK FORM 0.2 X+100 Y+90 Z+0
12 TOOL DEF 1 L+0 R+10
13 TOOL CALL 1 Z S4000
14 L Z+10 R0 F9999 M3
15 CALL LBL 1

Retract tool, end
16 L Z+50 M2
```



HEIDENHAIN

Advanced course TNC 4xx

C01



7288/3

Solution:

Milling cylindrical pins


SPGM

17 LBL 1
18 FN 1: Q23 = +Q3 + +2 Z-HEIGHT OF MOVEMENT
(CLEARANCE)
19 FN 4: Q24 = +Q4 DIV +2 RADIUS
20 FN 1: Q25 = +Q24 + +Q5 RADIUS + OVERSIZE
21 FN 1: Q34 = +Q25 + +Q108 RADIUS + OVERS. + TOOL RADIUS
22 FN 1: Q35 = +Q34 + +Q5 RAD. + 2 * OVERS. + TOOL RADIUS
23 FN 4: Q28 = +Q5 DIV +2 ONE HALF OVERSIZE
24 FN 4: Q27 = +Q7 DIV +2 ONE HALF FEED RATE
25 FN 3: Q29 = +Q7 * +2 DOUBLE FEED RATE
26 CC X+Q1 Y+Q2
27 LP PR+Q35 PA+0 R0 F9999 UNCORRECTED STARTING POINT
28 L Z+Q23 GO TO SETUP CLEARANCE
29 L Z-Q6 FQ27 PLUNGE
30 LP PR+Q25 PA+0 RL FQ7 MOVE TO WITH OVERSIZE
31 RND RQ28 FQ27 APPROACH TANG. WITH OVERS.
32 CP PA+0 DR- FQ7 CIRCLE WITH OVERSIZE
33 RND RQ28 FQ27 DEPART WITH OVERSIZE
34 LP PR+Q35 PA+0 R0 FQ29
35 LP PR+Q24 PA+0 RL FQ29 FINISH CUT
36 RND RQ28 FQ27
37 CP DR- FQ7
38 RND RQ28 FQ27
39 LP PR+Q35 PA+0 R0 FQ29 RETRACT TOOL
40 L Z+Q23 R0 F9999
41 LBL 0
42 END PGM 7288 MM



Task: **1 Bolt hole circle (parametric),
1 tool, 1 cycle (center)**

Program(s): _____



Q	Meaning	Value	Comment
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

SPGM 1

Starting values
Preparation

First hole

Start grinding

SPGM 1, end

<i>LBL 1</i>
<i>FN 0: Q24 = 1</i> <i>FN 0: Q25 = Q5</i> <i>FN 4: Q26 = 360 DIV Q4</i> <i>CC XQ1 YQ2</i> <i>LP PRQ3 PAQ5 R0 F9999 M3</i> <i>L Z+Q6 M99</i>
<i>LBL 2</i>
<i>FN 1: Q24 = Q24 +1</i> <i>FN 1: Q25 = Q25 +Q26</i> <i>LP PRQ3 PAQ25 M99</i> <i>FN 12: IF Q24 LT Q4 GOTO LBL 2</i>
<i>LBL 0</i>

Counter
Current angle
Angular step

Pole
Move to

Count
Calculate/update


Next holes

Check with
return jump



Solution:

1 Bolt hole circle (parametric), 1 tool, 1 cycle (center)



Main program

Center

Geometry values

Retract tool, end

```
0 BEGIN PGM 7416 MM
1 ..... 1 BOLT HOLE CIRCLE
2 BLK FORM 0.1 Z X+0 Y+10 Z-22
3 BLK FORM 0.2 X+100 Y+90 Z+0
4 TOOL DEF 1 L+0 R+3,5
5 TOOL CALL 1 Z S1000
6 L Z+10 R0 F9999 M3
7 CYCL DEF 1.0 PECKING ..... CYCLE TYPE AND CYCLE VALUES
8 CYCL DEF 1.1 SET UP -Q6
9 CYCL DEF 1.2 DEPTH -2
10 CYCL DEF 1.3 PECKG -2
11 CYCL DEF 1.4 DWELL 0
12 CYCL DEF 1.5 F100
13 FN 0: Q1 = +50 ..... X-CENTER
14 FN 0: Q2 = +50 ..... Y-CENTER
15 FN 0: Q3 = +20 ..... RADIUS
16 FN 0: Q4 = +10 ..... NUMBER
17 FN 0: Q5 = +0 ..... STARTING ANGLE
18 FN 0: Q6 = +2 ..... Z-SETUP CLEARANCE
19 CALL LBL 1
20 L Z+20 R0 F9998 M2 ..... RETURN JUMP
```

SPGM, Bolt hole circle

```
21 LBL 1
22 FN 0: Q25 = +Q5
23 FN 0: Q24 = +1
24 CC X+Q1 Y+Q2
25 FN 4: Q26 = +360 DIV +Q4
26 LP PR+Q3 PA+Q5 R0
27 L Z+Q6 M99

28 LBL 2
29 FN 1: Q25 = +Q25 + +Q26
30 FN 1: Q24 = +Q24 + +1
31 LP PR+Q3 PA+Q25 M99
32 FN 12: IF +Q24 LT +Q4 GOTO LBL 2
33 LBL 0
34 END PGM 7416 MM
```



HEIDENHAIN

Advanced course TNC 4xx


C05



7416/3

Task: **6 Bolt hole circles (parametrisch)**

Program(s): _____



Q	Meaning	Value	Comment
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

Program layout:
Intermediate step

**6 Bolt hole circles with 1 tool,
1 cycle, 1 clearance height**

**Structure for 1 tool /
1 cycle**

```
TOOL DEF 1 L0 R3,5
TOOL CALL ...
```

```
Q6 = +2      Set up
Q7 = +2      Depth
```

```
PECKING
SET UP -Q6
DEPTH -Q7
```

**Preparation for several
tools / cycles**

```
TOOL DEF 1 L0 R3,5
TOOL CALL ...
```

```
Q6 = +2      Set up
Q7 = +2      Depth
```

```
PECKING
SET UP -Q6
DEPTH -Q7
```

```
CALL LBL 10
```

Further steps / cycles

```
L Z100 F9999 M2
```

Geometry

1st bolt hole circle

```
Q1 = ... // Q5 = ...
```

```
CALL LBL 1
```

```
Q3 = ... /
```

2nd bolt hole circle

```
CALL LBL 1
```

```
Q... = / Q... =
```

3rd bolt hole circle

```
CALL LBL 1
```

```
⋮
```

6th bolt hole circle

```
CALL LBL 1
```

```
L Z100 M2
```

```
LBL 1
```

```
⋮
```

```
LP ... M3
L Z ... M99
```

```
LBL 3
```

```
⋮
```

```
LP PA M99
```

```
LBL 0
```

```
LBL 10
```

```
Q1 = ... // Q5 = ...
```

```
CALL LBL 1
```

```
Q3 = ... /
```

```
CALL LBL 1
```

```
Q... = / Q... =
```

```
CALL LBL 1
```

```
⋮
```

```
CALL LBL 1
```

```
LBL 0
```

```
LBL 1
```

```
⋮
```

```
LP ... M3
L Z ... M99
```

```
LBL 3
```

```
⋮
```

```
LP PA M99
```

```
LBL 0
```



HEIDENHAIN

Advanced course TNC 4xx


C01



74160/2

Solution:

6 Bolt hole circles with 1 tool, 1 cycle



Main program

```
0 BEGIN PGM 74160 MM
1 ..... 6 BOLT HOLE CIRCLES
2 BLK FORM 0.1 Z X+0 Y+0 Z-22
3 BLK FORM 0.2 X+100 Y+100 Z+0
4 TOOL DEF 1 L+0 R+3,5
5 TOOL CALL 1 Z S1000
6 L Z+10 R0 F9999 M3
7 FN 0: Q6 = +2
8 FN 0: Q7 = +2
9 CYCL DEF 1.0 PECKING
10 CYCL DEF 1.1 SET UP -Q6
11 CYCL DEF 1.2 DEPTH -Q7
12 CYCL DEF 1.3 PECKG -2
13 CYCL DEF 1.4 DWELL 0
14 CYCL DEF 1.5 F100
15 CALL LBL 10

Retract tool, end
16 L Z+50 R0 F MAX M2
```



HEIDENHAIN

Advanced course TNC 4xx

C05



74160/3

Solution:

6 Bolt hole circles with 1 tool, 1 cycle

SPGM, Bolt hole circle geometry

```
17 LBL 10
18 FN 0: Q1 = +50 ..... X-CENTER
19 FN 0: Q2 = +50 ..... Y-CENTER
20 FN 0: Q3 = +20 ..... RADIUS
21 FN 0: Q4 = +10 ..... NUMBER
22 FN 0: Q5 = +0 ..... STARTING ANGLE
23 CALL LBL 1 ..... 1st BOLT HOLE CIRCLE

24 FN 0: Q3 = +30
25 CALL LBL 1 ..... 2nd BOLT HOLE CIRCLE

26 FN 0: Q1 = +15
27 FN 0: Q2 = +15
28 FN 0: Q3 = +10
29 FN 0: Q4 = +8
30 CALL LBL 1 ..... 3rd BOLT HOLE CIRCLE

31 FN 0: Q1 = +85
32 CALL LBL 1 ..... 4th BOLT HOLE CIRCLE

33 FN 0: Q2 = +85
34 CALL LBL 1 ..... 5th BOLT HOLE CIRCLE

35 FN 0: Q1 = +15
36 CALL LBL 1 ..... 6th BOLT HOLE CIRCLE
37 LBL 0
```

SPGM, Bolt hole circle itself

```
38 LBL 1 ..... EXECUTION
39 FN 0: Q25 = +Q5
40 FN 0: Q24 = +1
41 CC X+Q1 Y+Q2
42 FN 4: Q26 = +360 DIV +Q4
43 LP PR+Q3 PA+Q5 R0
44 L Z+Q6 M99

45 LBL 2
46 FN 1: Q25 = +Q25 + +Q26
47 FN 1: Q24 = +Q24 + +1
48 LP PR+Q3 PA+Q25 M99
49 FN 12: IF +Q24 LT +Q4 GOTO LBL 2

50 LBL 0
51 END PGM 74160 MM
```



Program layout: **6 Bolt hole circles with 3 tools,
3 cycles, 2 clearance heights**

Preparation: *BLK- FORM*

Center

```
TOOL DEF 1 L0 R3,5
TOOL CALL ...
Q6 = +2
Q7 = +2
Q8 = +2
PECKING
SET UP -Q ...
DEPTH -Q ...
PECKG -Q ...

CALL LBL 10
```

Individual values:
SET UP
DEPTH
PLUNGING DEPTH
Cycle definition 1

SPGM–Aufruf

Pecking

```
TOOL DEF 2 L0 R2,5
TOOL CALL ...
Q7 = +20
Q8 = +10

CALL LBL 10
```

Individual values:
DEPTH
PLUNGING DEPTH

SPGM call

Tapping

```
TOOL DEF 3 L0 R3,0
TOOL CALL ...
Q... = +5
Q... = +15

TAPPING
SET UP -Q ...
DEPTH -Q ...

CALL LBL 10
```

Individual values:
SET UP
DEPTH

Cycle definition 2

SPGM call

Return jump *L Z100 M2*

Subprograms

```
LBL 10
Q1 = ... // Q5 = ...
CALL LBL 1
Q3 = ... // Q5 = ...
CALL LBL 1
:
LBL 0
```

```
LBL 1
:
LP ... M3
L Z ... M99
LBL 3
:
LP PA M99
LBL 0
```


Solution:

6 Bolt hole circles with 3 tools, 3 cycles

```
0 BEGIN PGM 74161 MM
1 ..... 6 BOLT H. C., 3 TOOLS, 3 CYCLES
2 BLK FORM 0.1 Z X+0 Y+0 Z-22
3 BLK FORM 0.2 X+100 Y+100 Z+0
Center 4 TOOL DEF 1 L+0 R+3,5
5 TOOL CALL 1 Z S1000
6 FN 0: Q6 = +2 ..... SET UP
7 FN 0: Q7 = +2 ..... DEPTH
8 FN 0: Q8 = +2 ..... PLUNGING DEPTH
9 L Z+10 R0 F9999 M3
10 CYCL DEF 1.0 PECKING
11 CYCL DEF 1.1 SET UP -Q6
12 CYCL DEF 1.2 DEPTH -Q7
13 CYCL DEF 1.3 PECKG -Q8
14 CYCL DEF 1.4 DWELL 0
15 CYCL DEF 1.5 F100
16 CALL LBL 10
17 STOP M6

Pecking 18 TOOL DEF 2 L+0 R+2,5
19 TOOL CALL 2 Z S3150
20 FN 0: Q7 = +20
21 FN 0: Q8 = +10
22 L Z+10 R0 F MAX M3
23 CALL LBL 10
24 STOP M6

Tapping 25 TOOL DEF 3 L+0 R+3
26 TOOL CALL 3 Z S400
27 FN 0: Q6 = +5 ..... SET UP
28 CYCL DEF 2.0 TAPPING
29 CYCL DEF 2.1 SET UP -Q6
30 CYCL DEF 2.2 DEPTH -15
31 CYCL DEF 2.3 DWELL 0
32 CYCL DEF 2.4 F400
33 L Z+10 R0 F MAX M3
34 CALL LBL 10
35 L M30
```



Solution:

6 Bolt hole circles with 3 tools, 3 cycles

Individual dimensions of bolt hole circles

36 LBL 10
37 FN 0: Q1 = +50 X-CENTER
38 FN 0: Q2 = +50 Y-CENTER
39 FN 0: Q3 = +20 RADIUS
40 FN 0: Q4 = +10 NUMBER
41 FN 0: Q5 = +0 STARTING ANGLE
42 CALL LBL 1

43 FN 0: Q3 = +30
44 CALL LBL 1

45 FN 0: Q1 = +15
46 FN 0: Q2 = +15
47 FN 0: Q3 = +10
48 FN 0: Q4 = +8
49 CALL LBL 1

50 FN 0: Q1 = +85
51 CALL LBL 1

52 FN 0: Q2 = +85
53 CALL LBL 1
54 FN 0: Q1 = +15
55 CALL LBL 1

56 L Z+50 R0 F MAX M5
57 LBL 0

Bolt hole circles

58 LBL 1
59 FN 0: Q25 = +Q5
60 FN 0: Q24 = +1
61 CC X+Q1 Y+Q2
62 FN 4: Q26 = +360 DIV +Q4
63 LP PR+Q3 PA+Q5 R0
64 L Z+Q6 M99


65 LBL 2
66 FN 1: Q25 = +Q25 + +Q26
67 FN 1: Q24 = +Q24 + +1
68 LP PR+Q3 PA+Q25 M99
69 FN 12: IF +Q24 LT +Q4 GOTO LBL 2

70 LBL 0
71 END PGM 74161 MM



Task: Linear hole pattern

Program(s): _____



Q	Meaning	Value	Comment
1	_____	_____	_____
2	_____	_____	_____
3	_____	_____	_____
4	_____	_____	_____
5	_____	_____	_____
6	_____	_____	_____
7	_____	_____	_____
8	_____	_____	_____
9	_____	_____	_____
10	_____	_____	_____
11	_____	_____	_____
12	_____	_____	_____
13	_____	_____	_____
14	_____	_____	_____
15	_____	_____	_____

Preparation

BLK FORM

Load data

*FN 0: Q1 = 10
 FN 0: Q2 = 10
 FN 0: Q3 = 9
 FN 0: Q4 = 12
 FN 0: Q5 = 10
 FN 0: Q6 = 7
 FN 0: Q7 = 2*

X start
 Y start
 X spacing
 Y spacing
 Num. of columns ||
 Number of rows =
 Z clearance

TOOL 1.../CYCL1... SET UP-Q7

CALL LBL 1

Retract tool, end

L Z50... M2

SPGM

LBL 1

Starting values

*FN 0: Q21 = Q1
 FN 0: Q22 = Q2
 FN 0: Q25 = 0
 FN 0: Q26 = 0
 FN 0: Q23 = Q3*

X current
 Y current
 Current column ||
 Current row =
 Copy X steps

Start grinding

LBL 2

Position

*L XQ21 YQ22 R0 F... M...
 L Z+Q7 M99
 FN 1: Q21 = Q21 + Q23
 FN 1: Q25 = Q25 + 1
 FN 12: IF Q25 LT Q5 GOTO LBL 2*

Pecking
 X step
 Count column

New row

*FN 0: Q25 = 0
 FN 1: Q26 = Q26 + 1
 FN 0: Q23 = -Q23
 FN 1: Q21 = Q21 + Q23
 FN 1: Q22 = Q22 + Q4
 FN 12: IF Q26 LT Q6 GOTO LBL 2*


Reset column counter
 Count row
 Reverse X step
 Compensation
 Y step

LBL 0



Solution:

Linear hole pattern with Q-parameters



Main program

Load data

Retract tool, end

```
0 BEGIN PGM 746910 MM
1 ..... LINEAR HOLE PATTERN
2 ..... WITH Q-PARAMETERS
3 BLK FORM 0.1 Z X+0 Y+0 Z-20
4 BLK FORM 0.2 X+100 Y+100 Z+0
5 FN 0: Q1 = +10 ..... X START
6 FN 0: Q2 = +10 ..... Y START
7 FN 0: Q3 = +9 ..... X SPACING
8 FN 0: Q4 = +12 ..... Y SPACING
9 FN 0: Q5 = +10 ..... NUMBER OF COLUMNS
10 FN 0: Q6 = +7 ..... NUMBER OF ROWS
11 FN 0: Q7 = +2 ..... Z CLEARANCE
12 TOOL DEF 1 L+0 R+3
13 TOOL CALL 1 Z S1400
14 L Z+50 R0 F MAX M3
15 CYCL DEF 1.0 PECKING
16 CYCL DEF 1.1 SET UP -Q7
17 CYCL DEF 1.2 DEPTH -10
18 CYCL DEF 1.3 PECKG -5
19 CYCL DEF 1.4 DWELL 0
20 CYCL DEF 1.5 F500
21 CALL LBL 1
22 L Z+50 R0 F MAX M2
```



HEIDENHAIN

Advanced course TNC 4xx

C05



746910/3

Solution:

Linear hole pattern with Q-parameters

SPGM

24 LBL 1 PREPARATIONS
25 FN 0: Q21 = +Q1 X
26 FN 0: Q22 = +Q2 Y
27 FN 0: Q25 = +0 COLUMN
28 FN 0: Q26 = +0 ROW
29 FN 0: Q23 = +Q3 X STEP

Start grinding

30 LBL 2
31 L X+Q21 Y+Q22 R0 F9999 POSITION
32 L Z+Q7 R0 F9999 M99 PECKING
33 FN 1: Q21 = +Q21 + +Q23 X STEP
34 FN 1: Q25 = +Q25 + +1 COLUMN
35 FN 12: IF +Q25 LT +Q5 GOTO LBL 2

Return jump

36 FN 0: Q25 = +0 COLUMN COUNTER
37 FN 1: Q26 = +Q26 + +1 ROW
38 FN 0: Q23 = -Q23 REVERSE X-DIRECTION
39 FN 1: Q21 = +Q21 + +Q23 COMPENSATION
40 FN 1: Q22 = +Q22 + +Q4 Y STEP
41 FN 12: IF +Q26 LT +Q6 GOTO LBL 2 NOT COMPLETE?
42 LBL 0
43 END PGM 746910 MM



HEIDENHAIN

Advanced course TNC 4xx

C01



746910/4